
1 Raison d’Maple

1.1 Four Properties of Maple

Here we intend to provide you with just enough information about the Maple language to give a
headstart at using it productively in the problem solving process. The particular version of Maple
that we are using as we describe it is Maple 9 running in Windows XP or 2000. You may be using
an earlier release on a different platform, but most of what is in this document is still relevant.

Maple has at least four properties which make it very useful in problem solving.

That means that you can type in commands and execute them, just like in the languages Basic
and Logo. A Maple command is simply a string of characters ending in a semicolon ’;’ or colon ’:’.
For example, the command

> 2+5, factor(x^2+5*x +6),expand((x+y)^2);

7, (x + 3) (x + 2), x2 + 2x y + y2

tells Maple to do a sequence of three things: add 2 and 5, factor the quadratic, and expand
the binomial. The strings factor and expand are called Maple ’words’. These are names of
procedures which have been defined for performing the action (sometimes) suggested by the name
on the stuff enclosed in parentheses just after the word. That stuff is usually called the ’input’ of
the word. The result of performing the procedure on the input is naturally called the ’output’ of
the word. If the command is terminated with a semicolon (colon), the output is displayed (not
displayed).

That means that it is built to work with algebraic expressions and draw pictures. There is a large
vocabulary of Maple words, such as factor, simplify, and expand which are used to ’symbolically
manipulate’ expressions in the manner you are used to doing with pencil and paper. There are also
a number of plotting words, which are used to draw graphs of functions of one or two variables,
curves and surfaces. These drawings can be animated (ie displayed in sequence) to study change.
Two of the plotting words most often used, plot and plot3d , are part of the regular Maple
vocabulary. The rest, and there are several, are found in the the plots package, a sort of specialized
vocabulary of words which is loaded separately.

What this means is that you can define additional words and add them to the vocabulary.
Initially, there will be little need to do this, except to define functions. The existing vocabulary is
large enough to carry out the solution to many problems. After awhile, it becomes very useful to be
able to add new words to the vocabulary. If you develop some words to work on a specialized class
of problems, these can be put into a package of words for easy access between worksheets. Maple
comes equipped with several such packages already, including plots , a package of drawing words,

linalg - a very useful package of vector and matrix manipulation words, combinat - a package of
words from combinatorics, and networks - a graph theory package.

We want to learn to use Maple to solve problems. In order make good use of the language
for this purpose, we need to become familiar with the worksheet environnment, get to know the
language, and foster our experimental urges.

1.2 The Worksheet: A handy place to solve problems.

When you click on the Maple Icon in Windows, an untitled worksheet opens up. Think of it as
a clean sheet of paper. Typically, after awhile, the worksheet will contain a record of the work
done to date on the problem or problems you have been working on. Very often, you might be in
a problem-solving team working on the problems. The worksheet can be given a name and saved
onto a disk for later working or for handing in. The worksheet file consists of a number of cells of
three different types: Input cells, Output cells, and Text cells.

These are started with a right angle bracket ’>’. Here are a couple.

An input cell is the place where you put the ’commands’ or ’statements’ you want Maple
to execute. The cell can contain one or more statements, each ending with a semicolon. The nice
thing about these cells in a worksheet is that they can be modified and reexecuted over and over
again. This enables you to correct typing mistakes with relative ease. For example, suppose I
wanted Maple to calculate ’3*(4+5*3)*(7+6);’ but left out a parenthesis.

> 3*(4+5*3)*(7+6 ;

Syntax error, ‘;‘ unexpected

An error message is generated which may help you find your mistake. So you can make a change
in that input cell and reexecute it. Use the mouse to put the cursor at the spot where the error
occurs and make the correction

You can also use this ability to change and reexecute input cells to change the numbers in
whatever problem you have worked out a solution to and see how the solution is affected.

Almost every input cell, when executed, gives an output cell containing the results of the
calculations. It is appended to the input cell which produced it. For example, let’s add 3 to the
23rd and 4 to the 12th – then let’s factor the result into prime factors.

> s := 3^23 + 4^12;

> ifactor(s);

s := 94159956043

(727) (129518509)

You inspect the output cell to see if it is what you want. If it is not, then go back and change
the input cell, and reexecute it.

Certain Maple words such as plot generate a separate window containing a picture or a page
of text (see factor in the next section). You can copy and paste these items into an output cell of
your worksheet if the need arises.

> plot(x^3-x+4, x=-2..2);

–2

0

2

4

6

8

10

–2 –1 1 2
x

Exercise: Execute the following plot command. Then copy the graph from the plot window
(use the Edit Menu in that window) and paste it into your worksheet.

> plot(x^3-x+4,x=-2..2);

–2

0

2

4

6

8

10

–2 –1 1 2
x

Text or Comment Cells:
A Text Cell is to record remarks and explanations of the solution to the problem you are

working on. All of the comments here are typed into text cells. It is a good idea to be liberal with
text cells. Undocumented calculations are for the most part worthless to anyone but the person
who made the calculations, and even then the value is a rapidly decreasing function of time.

You have very good control over the cells in your worksheet with the Menus at the top of this
worksheet. Depending on what platform you are using (Windows, X, or Mac), the menus may vary
in their titles, but the effects are the same with minor differences:

You change an input cell to a text cell and back again.
You can split a cell (input or text) and join cells.
You open up new cells (input or text) between existing cells.
You can erase cells (input, text, or output) at your discretion.
You can copy cells (input or text) and paste them into another location.
These controls make it relatively easy to work away on a worksheet, doing computations, record-

ing observations, explanations, etc and then go back and polish the worksheet up into a public
document.

The worksheet does not replace pencil and paper or the chalk board in the problem solving
process. When you are in the middle of analyzing a problem and deciding how to solve it, these
tools are extremely useful. It is easier to draw a rough diagram by hand than by Maple. After you
have decided on a plan and need to do some numerical calculation, some symbolic manipulation,
or some plotting to carry out the plan, then a knowledge of the Maple language becomes useful.

1.3 Get to know the language

Maple has a large built in vocabulary of words especially defined to carry out many of the algorithms
you have learned in your previous math classes. There are Maple words like factor , expand , simplify
, etc. You can learn about them with online Help .

There is a Browser available in both X and Windows Maple which has the Maple vocabulary
nicely indexed by category. Alternatively, you can ask for help in an input cell.

For example, to find out about factor just type
> ?factor

Knowing the word is one thing, but you also need to know the syntax of the word. What
is syntax? Every algorithm requires certain input information in order to be performed. After it
has been performed, certain output information is produced. To know the syntax of a Maple word
is to know the input information needed and the output information produced by the word. The
help screen gives you the syntax of the word, and thus tells you how to use it. For example, the
help screen for factor tells us (in CALLING SEQUENCES) that there are one and possibly two
inputs needed and one output produced by factor. The most useful part of the help screen is the
bottom part, where there are examples of the usage of the word in question. These examples can
be copied into a worksheet and tested out, which gives you a chance to develop a feel for the word
by experimenting with its use. In fact, there is a Maple word example which brings up a page of
examples of the usage of the word, in many cases.

1.3.1 Problems:

Exercise: Discover the difference between factor and ifactor using ?ifactor .

Exercise: Use ifactor to factor your social security number.

Exercise: Use nextprime to find the first 30 primes after 5000.

Exercise: How do collect , coeff and expand work? Use them to expand (x + 2 y)4 , collect the
result into a polynomial in x, and find the coefficient of the xˆ3 term of that result.

1.4 Experiment!

Experimentation is a prime source of learning. We are constantly conducting little experiments,
learning from them, and using that knowledge in some way. The same holds true when you
are learning Maple and/or working on some math problem. The help facility is a great aid to
experimentation. Say you are working on a math problem and you need to carry out some algorithm,
like solve an equation you have set up. You know the Maple word solve will carry out the algorithm,
but you have forgotten the syntax. You can use the word example to list some examples of its
usage, rather than bring up the entire help sheet on solve .

> example(solve);

Keep in mind that the help sheets are written for general use by both novices and experts, so
don’t be intimidated by unfamiliar terminology. Often the examples at the bottom of the sheet
suffice to tell you what you need to know. Keep using help!

1.4.1 Problems

Exercise: Spend a few minutes with the online help deciding what the terms list and set mean in
Maple. In particular, what is the main difference between a list and a set?

> example(list);

> example(set);

Exercise: If you haven’t already, find out the usage of the terms seq , NULL , subsop as used
in the example sheet for set and list.

Experiment is used at the beginning of solving a problem to generate data to conjecture a
solution. Towards the end of the process, experiment is used to test the solution. Maple makes the
act of experimentation easier to carry out and easier to modify. A Maple Worksheet makes it easier
to carry on work from one session to the next, and to prepare public documents, ie, homework
assignments, reseach papers, etc for public consumption.

2 An introduction to the Maple language

This section contains an introduction to some of the Maple vocabulary used for solving problems.
It is not meant to cover everything, just some of the basics. Read it through quickly, to get an
overview of the language. Then you can come back and read with more understanding later.

2.1 Arithmetic

First, there is arithmetic: addition, subtraction, multiplication, division and exponentiation. These
can be combined, just as on a calculator. The order of precedence is the the usual one: exponenti-
ation first, then multiplication and division, then addition and subtraction. So entering

> 2-3+4/5*6^7;

1119739

5

is the same as entering

> (2-3)+(4/5)*(6^7);

1119739

5

You will notice that Maple works with fractions whenever possible, changing to decimal numbers
only on demand. So typing and entering (pressing the enter key)

> 1/3 + 1/2;

5

6
will get a return of 5/6. If you put a decimal point in one of the numbers, that forces Maple to

return a decimal answer.
> 1/3. + 1/2;

0.8333333333

Another way to get decimals is to use the maple word evalf to convert a result to decimal form.
> evalf(1/2+1/3);

0.8333333333

Maple does arithmetic with complex numbers too. I is a Maple constant standing for
√
−1 . So

entering

> (3+2*I)*(2-I);

8 + I

will produce an output of 8+I .
The name for pi, the area of the circle of radius 1, in Maplese is Pi. So to calculate the area

of a circle of radius 3, you would enter

> Pi*3^2;

9π

2.2 Expressions, Names, Statements, and Assignments

Quantities to be computed like 1/2+1/3 are called expressions.
A name is a string of characters which can be used to store the result of a computation.
A statement in Maple is a string of names and expressions terminated with a semicolon, or

a colon if you don’t want to see the output, which when entered will produce some action.
The assignment statement is one of the most common statements. It is of the form
name := expression; For example, the assignment

> area := Pi*3^2;

area := 9π

stores 9*Pi in a location marked by the name area.
A more useful assignment for the area of a circle is

> area := Pi*r^2;

area := π r2

In this case, the expression Pi*rˆ2 is stored in area and with this assignment, the area of a
circle of any given radius can be computed using the Maple word subs. So to calculate the area
when r is 3, we enter

> subs(r=3,area);

9π

Here, it is convenient to think of the assignment as defining area as a function
of the radius r.

2.3 Functions

A function is a rule f (possibly very complicated) for assigning to each argument x in a given
set, a unique value f(x) in a set. In calculus the arguments and values of a function are always real
numbers, but the notion of function is much more flexible than that.

Functions can be defined in several useful ways in Maple.
As an expression: The assignment

> area := Pi*r^2;

area := π r2

defines the area of a circle as a function of it’s radius. The area function defined as an expression
is evaluated with subs. Since this function assigns real numbers to real numbers, its values can be
plotted on a graph with the Maple word plot. So the statement

> plot(area,r=0..4);

will produce in a separate plot window, the graph of the area function over the interval from
r=0 to r=4 .

With the arrow operator the assignment: If you have a simple function, you can often use the
arrow operator . For example,

> area := r -> Pi*r^2;

area := r → π r2

defines the area function also. Now to find the area of a circle of radius 3, we simply enter the
statement

> area(3);

9π

To plot this function over the domain r=0..4 , type

> plot(area,0..4);

0

10

20

30

40

50

1 2 3 4

Note that the variable r is omitted here.

Use unapply . The ugly little word unapply transforms expressions of one or more variables
into fuctions defined by an arrow operator. For example, if we had a polynomial defined by the
assignment

> pol := x^2 + 4*x -1;

pol := x2 + 4x − 1

then the assignment
> pol := unapply(pol,x);

pol := x → x2 + 4x − 1

turns pol into a function defined by an arrow operator.

As a procedure: The Maple word proc can be used to define functions. For example,

> area := proc(r) Pi*r^2 end;

area := proc(r)π ∗ r2 end proc

defines the area function too. It is evaluated and plotted as in the arrow operator definition.
One advantage of this way of defining a function is that the domain can be specified. For example,
the domain of the area function for a circle is all positive real numbers. This can be inserted into
the procedure, with the Maple word ERROR .

The message must be enclosed in backquotes ’‘’, which is on the key with the tilde .

> area := proc(r)

> if r <= 0 then ERROR(‘radius must be positive‘) else

> Pi*r^2 fi end;

area := proc(r)

if r ≤ 0 thenERROR(‘radius must be positive‘) else π ∗ r2 end if

end proc
> area(3);

9π

> area(-3);

Error, (in area) radius must be positive

Note the if..then..fi control statement here.
You can learn more about the word if by typing ?if in an input cell and entering it.
Functions of two variables can be defined and plotted just as easily in Maple as functions of

one variable. For example, the volume V of a cylinder of height h and radius r is defined by
> V := (r,h) -> Pi*r^2*h;

V := (r, h) → π r2 h

To see what the graph of V looks like, use plot3d .

> plot3d(V,0..4,0..4,axes=boxed);

Which way of defining a function is the preferred way? That really depends on the situation. The
expression method works well for functions which have only one rule of evaluation, but eventually
you cannot avoid using an -> or proc definition. You will find yourself using arrow or proc definitions
more and more as time goes by.

Piecewise defined functions:
Many functions can only be described by stating various rules for various parts of the domain.

The Maple word piecewise will help with defining such functions.
Here is an example to show usage.
> f(x) :=piecewise(x <= -1,x^3+8, x <= 2,7+ 2*x, x <= 4, 11 -

cos(x),3*x);

f(x) :=















x3 + 8 x ≤ −1
7 + 2x x ≤ 2
11 − cos(x) x ≤ 4
3x otherwise

> f(2);

f(2)

As it stands, f is not really a function. We need to use unapply to make it into a funtion.

> g :=unapply(f(x),x);

g := x → piecewise(x ≤ −1, x3 + 8, x ≤ 2, 7 + 2x, x ≤ 4, 11 − cos(x), 3x)

> g(2);

11

When plotting piecewise defined functions, sometimes style = point is better.

> plot(g, -3..6,style= point);

2.4 Built in Maple functions and Operations with Functions

All of the standard scientific functions are built into Maple. For example, sqrt is the square root
function, abs is the absolute value function, the trig and inverse trig functions are sin , arcsin , cos
, etc., the natural logarithm and exponential functions are ln and exp . For a complete list of built
in functions, type

> ?inifcns;

New functions can be obtained from old functions by use of the arithmetic operations of addi-
tion, subtraction, multiplication, and division together with the operation of composition, which is
denoted by @ . Thus the function defined by the assignment

> y := sin(cos(x^2+3));

y := sin(cos(x2 + 3))

and evaluated at x=3 by
> subs(x=3.,y);

sin(cos(12.))

could also be defined by the assignment
> y := sin@cos@(x->x^2+3);

y := sin@cos@(x → x2 + 3)

and evaluated at x=3 by
> y(3.);

0.7472099577

2.5 Using Maple as a fancy graphing calculator.

It is convenient to think of Maple as a fancy graphing calculator for many purposes. For example,
suppose you want to find the real solutions of the equation x5 − 30x − 2 = 0 in the interval −3..3
. Then we can just plot the right hand side of the equation and look for where the graph crosses
the x-axis.

> f := x -> 10*x^5 - 30*x +10 ;

f := x → 10x5 − 30x + 10

> plot(f,-3..3);

By inspection, the graph crosses near 0. We can look closer.

> plot(f,-1.5..1.5);

We see that the graph crosses 3 times, the largest solution being between 1 and 1.5. If we
wanted the largest solution more accurately, we could use fsolve. Note the syntax. There are three
arguments, the equation to solve, the variable to solve for, and the interval in which to search for
a solution.

> fsolve(f(x)=0,x,1..1.5);

1.214648043

2.6 Data types, Expression Sequences, Lists, Sets, Arrays, Tables:

Maple expressions are classified into various data types . For example, arithmetic expressions are
classified by whether they are sums type ’+’ , products type ’*’ , etc.

The Maple word whattype will tell what type a particular expression is.

> whattype(1/2);

fraction

> whattype(a + b);

+

> whattype(x^2 + x = 2*x - 1);

=

> whattype(a,b,3);

exprseq

Expression Sequence.
An exprseq , expression sequence, is any sequence of expressions separated by commas. For

example,

> viola := 1,2, w*r+m, a=b+c, 1/2, (x+y)/z,‘hello‘;

viola := 1, 2, w r + m, a = b + c,
1

2
,

x + (sin@cos@(x → x2 + 3))

z
, hello

is an assignment to viola of an expression sequence of 7 expressions. To refer to the sixth
expression in this sequence, use the expression viola[6];

> viola[6];

x + (sin@cos@(x → x2 + 3))

z

List.
A list is an expression sequence enclosed by square brackets. So
> explist:= [viola];

explist := [1, 2, w r + m, a = b + c,
1

2
,

x + (sin@cos@(x → x2 + 3))

z
, hello]

makes a list whose terms are those in viola . As with expression sequences, we can refer to
particular terms of a list by appending to its name the number of the term enclosed in square
brackets. Thus to get the fifth term of explist , type the expression

> explist[3];

w r + m

You can also reference the fifth term in this list by by using the Maple word op .
> op(3,explist);

w r + m

In general, op(n,explist); returns the nth term in the list explist .
To count how many terms are in a list, use the word nops . So for example,

> nops(explist);

7

tells us that there are 7 terms in the list explist . nops comes in handy when you
don’t want to (or aren’t able to) count the terms in a list by hand.

You can’t directly use the word nops to count the number of terms in an expression sequence.
But you can put square brackets around the expression sequence and count the terms in the resulting
list. This device is used again and again.

> nops(3,4,a);

Error, wrong number (or type) of parameters in function nops

> nops([3,4,a]);

3

A point in the plane is a list of two numbers. Points can be added and subtracted and
multiplied by a number.

> p := [1,2]; q := [-3,1];

p := [1, 2]

q := [−3, 1]

> w := 3*p + 2*q - p;

w := [−4, 6]

One important use of lists is to make lists of points to plot. For example, to draw a picture of
the square with vertices (1,1), (3,1), (3,3), (1,3), make a list and then plot it.

> ab := [[1,1],[3,1],[3,3],[1,3],[1,1]];

ab := [[1, 1], [3, 1], [3, 3], [1, 3], [1, 1]]

> plot(ab);

Notice in the graph that the origin is not included in the field of view. We can specify that by
restricting the x and y coordinates.

> plot(ab,x=0..4,y=0..4);

Error, (in plot) invalid arguments

Another use of lists is with parametric plots . If you have a curve in the
plane described parametrically with x = f(t), y = g(t) , as the parameter t runs from a to b,

then you can draw it by making up a 3 term list to give to plot. Say you wanted to draw the upper
half of the circle of radius 4 centered at (1,5). Then the list consists of the expressions for the x
and y coordinates followed by an equation giving the range of the parameter.

> plot([1+4*cos(t),5+4*sin(t),t=0..Pi],
scaling=constrained);

If you had to draw several pieces of circles, you might define a function to simplify things. You
can call the function whatever you want, say circ.

> circ := (h,k,r,f,l) -> [h+r*cos(t),k+r*sin(t),t=f..l];

circ := (h, k, r, f, l) → [h + r cos(t), k + r sin(t), t = f..l]

So if we wanted circles of radius 1/2 centered at the corners of the square ab we can construct
the sequence of lists

> circs := seq(circ(op(ab[i]), 1/2,0,2*Pi),i=1..4);

circs := [1 +
1

2
cos(t), 1 +

1

2
sin(t), t = 0..2π], [3 +

1

2
cos(t), 1 +

1

2
sin(t), t = 0..2π],

[3 +
1

2
cos(t), 3 +

1

2
sin(t), t = 0..2π], [1 +

1

2
cos(t), 3 +

1

2
sin(t), t = 0..2π]

In order to plot these circles, you need to enclose them in curly brackets to make a set of the
sequence before you give them to plot . See below for a discussion of sets.

> plot({circs,ab},scaling=constrained);

Sometime you might want to split a list of points to plot into a list of x-coordinates and another
list of ycoordinates. The Maple word seq is very handy for this and many other operations. So
to split off from ab the odd and even terms–

> xdat := [ seq(ab[i][1],i=1..nops(ab) )];

xdat := [1, 3, 3, 1, 1]

> ydat := [seq(ab[i][2],i=1..nops(ab) )];

ydat := [1, 1, 3, 3, 1]

What about the converse problem? Building up a list of points to plot from two lists can also
be done. The first thing you might think of doesn’t work, however.

> seq([xdat[i],ydat[i]],i=1..nops(xdat));

[1, 1], [3, 1], [3, 3], [1, 3], [1, 1]

Seq doesn’t work well with a pure expression sequence as input. However, with some coaxing
we can get it to do what we want.

> newab :=[seq([xdat[i],ydat[i]],i=1..nops(xdat))];

newab := [[1, 1], [3, 1], [3, 3], [1, 3], [1, 1]]

What did we do to change the input to seq ? We enclosed it in square brackets. If you feed such
a list of points to plot, it knows what to do. If you wanted to strip out the inside brackets, that
can be done too, but in release 4 of Maple, plot would treat it as a sequence of constant functions.

> newab := [seq(op([xdat[i],ydat[i]]),i=1..nops(xdat))];

newab := [1, 1, 3, 1, 3, 3, 1, 3, 1, 1]

> plot(newab,color=black);

Sets
A set is an expression sequence enclosed by curly brackets. This is much different from a list.

For one thing, the order in which you specify the members of a set may not be the order in which
they are stored. Also each member of the set is only stored once, no matter how many times you
list it.

> Aset := {y+x+1,1,2,1,4,‘bill‘,x+y+1,‘bill‘};
Aset := {1, 2, 4, bill , (sin@cos@(x → x2 + 3)) + x + 1}

The set operations of union, intersection , and minus are at your beck and call.

> Anotherset := Aset union {4,3,a,7} ;

Anotherset := {1, 2, 3, 4, 7, a, bill , (sin@cos@(x → x2 + 3)) + x + 1}
> Anotherset minus Aset, Anotherset intersect Aset;

{3, 7, a}, {1, 2, 4, bill , (sin@cos@(x → x2 + 3)) + x + 1}
Sets are important when plotting more than one function at at time, to plot the quadratic

function x2 − 2 and the linear function 2x + 5 on the same axes,
> plot({x^2-2,2*x+5},x=-5..5);

plots the parabola y = x2 − 2 and the line y = 2x + 5 over the domain x = −5..5 on the same
graph. If you have a very complicated drawing to make, you can use plots[display] from the
plots package. Just give names to the plots you want to display and then display the list of plots
you have named.

> pl1 := plot({x^2-2,2*x+5},x=-5..5):
> pl2 := plot([[2,1],[3,20],[0,0],[2,1]]):

> plots[display]([pl1,pl2]);

Tables and Arrays
A table is a special kind of data structure which is very flexible. The packages of special

vocabularies are really tables whose indices of the package are the names of the procedures and
whose entries are the bodies of the procedures. We do not make much use of tables in this handbook,
except for arrays.

An array is a special kind of table whose indices are numerical. Somet useful arrays are
matrices (2 dimensional arrays) and vectors (1 dimensional arrays).

Matrix operations are made using Maple word evalm together with the symbol for matrix
multiplication &* .

> a := array(1..2,1..2);

a := array(1..2, 1..2, [])

creates a 2 by 2 matrix, whose entries are accessed as a[1,1] etc.
So to rotate the square ab := [[1, 1], [3, 1], [3, 3], [1, 3], [1, 1]] through an angle of 31 degrees

counter clockwise about the origin and display it, we could proceed as follows.
> rot := array([[cos,-sin],[sin,cos]]);

rot :=

[

cos −sin
sin cos

]

> ang := evalf(Pi/180*31);

ang := 0.5410520681

> ab := [[1,1],[3,1],[3,3],[1,3],[1,1]];

ab := [[1, 1], [3, 1], [3, 3], [1, 3], [1, 1]]
> rotab := [seq(convert( evalm(rot(ang)&*ab[i]),list) ,i=1..nops(ab)

)];

rotab := [[0.3421292258, 1.372205376], [2.056463827, 2.402281526],

[1.026387677, 4.116616127], [−0.6879469243, 3.086539977],

[0.3421292258, 1.372205376]]

> plot({ [[0,0]],ab,rotab} );

2.7 Maple control statements

There are two especially important control statements . One is the repetition loop , and the
other is the conditional execution statement. The repetition loop is

for .. from .. by .. to .. while .. do .. od;

This statement can be used interactively or in a procedure to perform repetitive tasks or to do
an iterative algorithm.

Example: Add up the first 100 numbers.

> s := 0: for i from 1 to 100 do s := s+i od:

> s;

5050

Example: Compute the cubes of the first five positive integers and store them in a list. Then
do it again, storing them in an array.

Solution with lists:
> locube := NULL: # start with the empty exprseq

for i from 1 to 5 do
locube := locube ,i^3 od:
locube := [locube]; # make locube a list.;

locube := [1, 8, 27, 64, 125]

Note the way the list is built up from an empty exprseq NULL . Each time through the loop,
one more term is added onto the end of the sequence. At the end, square brackets are put around
the sequence, making it a list. With arrays, one can be more direct.

Solution with arrays:
> aocube := array(1..5): # initialize the array.

> for i from 1 to 5 do aocube[i]:= i^3 od;

aocube1 := 1

aocube2 := 8

aocube3 := 27

aocube4 := 64

aocube5 := 125

> op(aocube); # to see the array

[1, 8, 27, 64, 125]

Now the array aocube has the numbers stored in it. To refer to the third element of aocube ,
we would enter aocube[3] just as if it were a list, rather than an array. Why have arrays at all?
Well, for one thing, the terms in an array can be more easily modified. For example, to change the
third term in aocube to 0 just enter aocube 3 := 0; . To change the third term in locube to 0, you
have to make an entirely new list whose terms are all the same as locube except for the third one.

> aocube[3]:=0;

aocube3 := 0

> print(aocube);

[1, 8, 0, 64, 125]

> locube := [locube[1],locube[2],0,locube[4],locube[5]];

locube := [1, 8, 0, 64, 125]

Conditional execution

if .. then .. elif .. else .. fi;

There are lots of times when you need to consider cases, and they can all be handled with the
if .. then .. elif .. else .. fi; statement. For example, many functions are defined piecewise.

The absolute value function abs is such a function.
Problem: Define your own version of the absolute value function.

A solution:

> myabs := proc(x) if x > 0 then x else -x fi end;

myabs := proc(x) if 0 < x thenx else − x end if end proc

> myabs(-23);

23
> plot(myabs,-2..2,scaling=constrained,title=‘my absolute value‘); #

to see what it looks like.

2.8 A Brief Vocabulary of Maple Words

Here are some Maple words useful in calculus problem solving, together with examples of their
usage. For more information on these words and others, look at the helpsheets and use the help
browser.

> y := (x+3)/tan(x^2-1); # use ’colon-equal’ to make assignments.

y :=
x + 3

tan(x2 − 1)
> collect(x*2 + 4*x,x); # collects like powers of x.

6x

> diff(cos(x),x); # calculates the derivative

−sin(x)

> D(cos); # the differential operator

−sin

> y := denom((a+b)/(e+f)); # assigns e+f to y.

y := e + f

> y := ’y’; # makes y a variable again.

y := y

> evalc((2+3*I)^3); # performs complex arithmetic

−46 + 9 I

> evalf(1/2^9); #evaluates 1/2^9 to a decimal number

0.001953125000

> expand((x+b)^7); # expands the product

x7 + 7x6 b + 21x5 b2 + 35x4 b3 + 35x3 b4 + 21x2 b5 + 7x b6 + b7

> p := x^2+5*x+6; # assigns the quadratic to p.

p := x2 + 5x + 6

> factor(p); # factors the polynomial

(x + 3) (x + 2)

> fsolve(x^5-3*x=1,x,0..2); # solve eqn for x in 0..2

1.388791984

> int(x*exp(x),x); # returns an antiderivative.

x ex − ex

> Int(x*exp(x),x=0..1); # A passive integral.
∫ 1

0
x ex dx

> map(x->x^2,[1,3,2,5]); # returns a list of squares.

[1, 9, 4, 25]

> nops([3,4,x,1]); # returns the number of terms in the list.

4

> numer((a+b)/c); # gives numerator, here a+b

a + b

> op([3,4,1,x]); # strips the brackets off the list

3, 4, 1, x

> plot(x^2+x, x=-3..3); # plots x^2+x as x goes from -3 to 3.

> plot3d(x^2+y,x=-2..2,y=0..2); # plots a surface

> f := x -> x^2; # defines the squaring function.

f := x → x2

> f(3); # then returns 9.

9

> quo((x^4-4),(x^2-2),x); # divides polynomials

x2 + 2

> iquo(23,2) ; # divides the integers

11

> rem((x^4-4*x+3),(x^2-2),x); # gives the remainder

7 − 4x

> irem(23,2) ; # gives the integer remainder

1

> restart; # very handy. This word resets all assignments.

> eq1 := x^2 + 3*x -1 = a; # assigns the equation

eq1 := x2 + 3x − 1 = a

> rhs(eq1); # yields the righthand side of eq1. There is also an lhs.

a

> simplify(a/x+b/y); # sometimes simplifies expr.

a y + b x

x y
> solve(a*x+4*y=0,x); # solve the equation for x.

−4 y

a
> subs(x=5,x^2+x); # substitute 5 for x where it occurs in x^2+x.

30

> i := ’i’; # makes i a variable again

i := i

> sum((i^2,i=2..9)); # add up the 2nd thru 9th squares

284

2.9 Trouble Shooting Notes

Learning to use Maple can be an extremely frustrating experience, if you let it. There are some
types of errors which occur from the beginning that can be spotted and corrected easily by a person
fluent in Maple, so if you have access to such a person, use him or her.

Here are a few suggestions that may be of use when you’re stuck with a worksheet that’s not
working like it should.

• Use help: There is a help sheet with examples for every Maple word. A quick read thru
will often clear up syntax problems. One very common early mistake is to leave out the
parentheses around the inputs of a word. For example, typing

> plot x^2;

Error, missing operator or ‘;‘

will get you a syntax error, because you left out the parentheses.

• The maple prompt is ‘>‘ . You can begin entering input after it. Make sure you are typing
into an input cell, if you are expecting output.

• End maple statements with a semicolon ‘;‘ . Maple does nothing until it finds a semicolon. If
you are getting no output when you should be, try feeding in a semicolon. This often works.

• When in doubt, put in parentheses. For example, (x+3)/(x-3) is very different from x+3 /
x-3 .

• Make sure your variables are variable. You may have assigned a value, say 3, to x in a previous
problem. To make x a variable again, type x := ’x’: . Use the forward quote ’ key, just below
the double quote ” here. If you forget this, strange things can happen. One way to handle
this is to keep an input cell of variables used.

• Use restart; By typing restart; in an input cell and pressing enter, you clear all assignments,
and start with a clean slate. This fixes a lot of problems fast, but you will need to re-execute
input cells.

• Are you using the correct quote symbol? In Maple, the forward quote ’ is used to suppress
evaluation. The back quote ‘ is used to enclose ascii strings. The double quote ” is used to
reference the last computation.

• Do not forget to end loops with od, ‘if‘ statements with fi, and procedures with end. If you
start a loop with do , Maple does not begin processing until it finds the end of the loop,
which is signaled by the word od; The same applies to the if .. then ... fi; and proc ... end;
contructions. If you are getting no output when you should be, try feeding an od; , fi; , or
end; This often works.

• Unwanted output?: Is there output you need but don’t want to see? Use a colon ‘:‘ instead
of a semicolon to end the Maple statement which generates the output.

• Use printlevel := 10; if you want to see what Maple is doing behind the scenes when you give
it a command. If you want to see more, use printlevel := 50 or higher. Often by inspecting
the output when printlevel is greater than 1 (the default), you can discover what is ailing
your worksheet.

• Use debug. If you have defined a word, say ‘something‘ and it does not do what you want,
you can often discover the error by typing debug(something); in an input cell and pressing
the enter key. When you use the word again, its behind the scene computations are printed
out for your inspection.

• Want to see a word definition? Say you want to see how plot works. Type interface(verboseproc=2);
in an input cell and press enter. Then type print(plot);

3 Setting Up and Solving Problems

3.1 What is a problem?

For our purposes, a problem is anything that can be formulated as a question whose answer involves
some mathematics. The main use of mathematics is to solve problems, and the best way to learn
mathematics is to solve problems. This idea of a problem includes the ’skill’ exercises that are
always at the end of the sections in mathematics textbooks. For example, early in beginning
algebra there is a section on solving linear equations, and at the end of that section there is a set
of exercises consisting of lots of problems like ’solve 4x + 3 = 2x − 6 for x’.

It also includes ’word problems’ which can be solved using the methods mastered in the skill
exercises. The word problems often are stated at the beginning of the section as motivation for the
methods which are developed in the section. Problems usually arise in a context. Once the context
is well understood, the problem can be formulated or posed and a method of solution worked out.
From this solution, other problems may arise which require solving. We want to consider this
problem identification and formulation as part of problem solving also.

The process of solving a problem is an active process, but can get bogged down for lack of
knowing what to do next. So it is helpful to have a list of things to do. Here is one list of steps to
carry out when you are solving a problem.

3.2 Setup – Solve – Interpret

SETUP the problem. This involves several steps.

• Pose/Read the problem carefully, getting straight the meaning of all the terms used in the
statement.

• Draw a picture or diagram. This is a good way to focus thoughts, and gives you a place to
put your labels.

• Label or list the dimensions (or variables) important to the problem. Among these are
the given dimensions, that is, the dimensions whose values are stated in the problem, the
requested dimensions, that is, the dimensions whose values are requested in theproblem, and
the intermediate dimensions, that is, the dimensions which arise in the process of trying to
determine the requested dimensions from the given dimensions.

• List or derive the equations relating the labeled dimensions.

SOLVE the equations we have set up for the required dimensions in terms of the given
dimensions. This is one of places where Maple comes in very handy. It is easy to get bogged down
in the calculations so that you lose all interest in solving the problem. This is less likely to happen
if you have Maple at your disposal.

INTERPRET the obtained solutions. What are the realistic solutions given the context
of the problem? Which should be ignored? Have all solutions been obtained?

Where does Maple come in handy in this process? In the Solve phase, mostly. The actual
setting up of the problem as a mathematical problem has to be done by you. Regard Maple as
a tool to carry out and record the solution you imagine. A very important word in the Maple
vocabulary is solve. This word is used to solve a system of one or more equations which come up.

3.3 A Swimming Pool Problem:

Problem: A swimming pool is three times as long as it is wide. It is also 40 feet longer than it is
wide. Find its dimensions.

Solution: Let l and w be the length and width of the pool. Then the first statement of the
problem translates to the equation l = 3w , and the second statement to l = w + 40 . We need to
solve these two equations simultaneously for l and w. First, set up the equations.

> eq1 := l = 3*w;

eq1 := l = 3w

> eq2 := l = w + 40;

eq2 := l = w + 40

Then solve the system for l and w. We can do this by subtracting eq2 from eq1 and solving for
w, getting w = 20 , then substituting that value for w into eq1 getting l = 60 .

> eq3 := eq1 - eq2;

eq3 := 0 = 2w − 40

> eq4 := lhs(eq3) - 2*w = rhs(eq3) - 2*w;

eq4 := −2w = −40

> eq5 := -(1/2)*eq4;

eq5 := w = 20

> eq6 := subs(eq5,eq1);

eq6 := l = 60

> solution := {eq5,eq6};
solution := {w = 20, l = 60}

The word solve carries out this algorithm automatically.

> solve({eq1,eq2},{l,w});
{w = 20, l = 60}

The pool is 20 feet wide and 60 feet long.
The word solve is a very good to know, but it is not infallible. It uses some methods of solving

equations which you know and some which you probably don’t know. Don’t give up just because
solve doesn’t give you a solution.

3.4 Four methods of solving equations

So it is important to also be able to solve equations by various means. Here are four.
Guess and check. This method involves guestimating a solution somehow and checking your

accuracy somehow. For example, suppose you had guestimated that the dimensions of the pool are
about 25 by 55 feet and wanted to check that. Use the word subs.

> subs({w=25,l=75},[eq1,eq2]);

This line says to substitute 25 and 75 for w and l in eq1 and eq2.
(Notice the use of braces and brackets here. Braces are used to enclose the members of a set

and brackets are used to enclose the members of a list. In a set, the order is not important and
repetitions are not counted; in a list, the order is important and repetitions are counted.) As we
can clearly see, our guestimate is off. We have satisfied the first equation, but not the second. If
we decrease the value of w by 1, then we have to decrease the value of l by 3 in order to continue
to satisfy the first equation. Will we come closer to satisfying the second? Let’s see.

> subs({w=24,l=72},[eq1,eq2]);
[75 = 75, 75 = 65]

[72 = 72, 72 = 64]

Well, yes, if ever so slightly. The left-hand side of the second equation has decreased by 2
towards the right-hand side. We could pursue this method of guessing and then trying to improve
our guess, but let’s put that off until later.

By Hand, as with pencil and paper.
You can choose to try to solve your equations by hand by which I mean to manipulate the

equations the same way you would do using pencil and paper. This can be done without using
solve at all. This is what we did in our original solution to the swimming pool problem. On the
other hand, we could simplfy the equations and then use the word solve. For example, let’s solve
the two equations, eq1 and eq2, by hand. Using the Maple word solve we need only solve one
equation for one unknown. Here is a possible way to proceed:

> eq3 := subs(l=solve(eq1,l),eq2);

eq3 := 3w = w + 40

> sol1 := w = solve(eq3,w);

sol1 := w = 20

> sol2 := l = solve(subs(sol1,eq1),l);

sol2 := l = 60

> sol := {sol1, sol2};
sol := {w = 20, l = 60}

Graphical solution.
Another way to solve equations is graphically. Here we can plot each equation, using plot or

implicitplot and use the pointer to locate the approximate solutions. These solution(s) are the
located where the graphs of the equations coincide. Note: the word implicitplot can only be found
in Maple V Release 2. Earlier versions of Maple only plot functions. This method also works very
well on a graphing calculator.

> with(plots):

> implicitplot({eq1,eq2},w=10..30,l=50..70);

50

55

60

65

70

l

10 15 20 25 30
w

Using fsolve.
Another way to solve equations is with fsolve . This word employs methods of calculus to find

floating point approximations to the equation you are trying to solve. If you do not supply fsolve
with an interval in which to search for a solution, then it returns the first solution it finds. This
word works well in conjunction with plot . You can use plot to narrow down the search interval,
and then use fsolve to get the ’exact’ answer.

3.5 Problems

A. Solve the following systems of equations.
1. x2 + y2 = 10 , y = 3x

2. a + 3 b + c = 10, 2 a − b + 72 c = 20, 7 a − 3 b + 21 c = 30 .

3. x5 − 5x2 + 2 = 0 use plot and fsolve here.

4. x − 3 y + z = a, 2x + y − 3 z = b, 9x + 3 y + 5 z = c for x, y and z.

B. Solve the following word problems by setting up and solving a system of equations.

1. The height of the Eiffel Tower in Paris is 125 feet less than twice the height of the Washington
Monument. The latter is 75 higher than the Great Pyramid of Cheops and 105 feet higher than
the dome of St. Peter’s Church in Rome. If the sum of the heights of these four edifices is 2510
feet, find the height of each to the nearest foot.

2. A large conference table is to be constructed in the shape of a rectangle with two semicircles
at the ends (see figure). Find the dimensions of the table, given that the area of the rectangular
portion is to be twice the sum of the areas of the circular ends, and the perimeter of the whole
table is to be 40 feet.

Here is a diagram to accompany the problem.
> F:=plot([-cos(t)+1,sin(t)+1,t=Pi/2..-Pi/2]):

> G:=plot([cos(t)+6,sin(t)+1,t=Pi/2..-Pi/2]):

> H:=plot([[1,0],[6,0],[6,2],[1,2],[1,0]],style=LINE):

> plots[display]({F,G,H},axes=none,scaling=constrained);

3. A local hardware store worker is making up a fertilizer mix from some left over fertilizer from
the summer. There are three types, with 30 percent, 20 percent and 15 percent nitrogen content
respectively. When he mixed all the fertilizers together and tested the nitrogen content, he found
that the mix weighed 600 pounds and contained 25 percent nitrogen content. How many pounds
of each type did he have left over? How many solutions does this problem have?

4. A 30 foot ladder and a 40 foot ladder are positioned so as to cross each other in an alley.
That is, they are leaning up against opposite walls with their bases snug up against the base of the
opposite walls. Given that they cross at a point 10 feet above the floor of the alley, determine the
width of the alley.

5. Make up your own algebra problem to set up and solve using Maple. It can be a variation
on one of the problems above, or it can be something entirely different.

3.6 More About Plotting

We have already used a few words from the with(plots); package. When you want to learn more
about any package (in this case with(plots);) start by looking at the commands in the package. Do
this by typing

> with(plots);

[animate , animate3d , changecoords , complexplot , complexplot3d , conformal , contourplot ,

contourplot3d , coordplot , coordplot3d , cylinderplot , densityplot , display ,

display3d , fieldplot , fieldplot3d , gradplot , gradplot3d , implicitplot , implicitplot3d ,

inequal , listcontplot , listcontplot3d , listdensityplot , listplot , listplot3d , loglogplot ,

logplot , matrixplot , odeplot , pareto , pointplot , pointplot3d , polarplot , polygonplot ,

polygonplot3d , polyhedraplot , replot , rootlocus , semilogplot , setoptions ,

setoptions3d , spacecurve , sparsematrixplot , sphereplot , surfdata , textplot ,

textplot3d , tubeplot ]

We have already used plot , implicitplot , and display (in problem 2 of part B). To learn more
about these very useful commands simply type

> ?plot

> ?implicitplot

> ?display

> ?textplot

> ?animate

or anything else that might interest you. Look at the bottom of the help files for examples.
Sometimes, you are your best teacher. Learn by experimenting with a few (or all) of these com-
mands. Also, you will learn more about these words in worksheets to come.

3.7 Putting in a parameter.

One of the advantages of solving a problem in a Maple worksheet is that it gives the capability
of going back and changing the numbers in the problem to study how the solution changes. In
fact, you are led to the practice of putting a parameter into the problem. For example, to put a
parameter in the swimming pool problem, we have two natural choices: Replace the number 3 or
the number 40 by a parameter. Let’s replace 3 with p.

Parameterized Swimming Pool Problem

Problem: swimming pool is p times as long as it is wide. It is also 40 feet longer than it is
wide. What are its dimensions in terms of p? Describe how the dimensions vary with p.

Solution: We could just go back to the cell containing the original equations and put in a p
for 3 in eq1.

> restart;

The word restart clears all variables by rebooting Maple.
> p := ’p’;

p := p

> eq1 := l = p*w;

eq1 := l = pw

> eq2 := l = w + 40;

eq2 := l = w + 40

Now we would solve for l and w, just as before, except now the solution is given in terms of the
parameter p.

> sol := solve({eq1,eq2},{l,w});

sol := {l = 40
p

−1 + p
, w =

40

−1 + p
}

Now by inspection, we can see that as p gets large, both w and l get small. Also, as p approaches
1 from the left, l and w get large. In more complicated situations, we could use plot to study how
the solution changes as the parameter changes.

> plots[animate]({p*w,w+40},w=0..80,p=0..2);

The word animate, which is used out of the with(plots); package, is very useful in seeing the
range of possible solutions for this problem. Press your pointer on play to see the animation.

3.8 Problems

1. What happens to the solution to the swimming pool problem as the difference of the width and
length is allowed to vary? (Keep the ratio of the width to the length constant.)

2. A large conference table is to be constructed in the shape of a rectangle with two semicircles
at the ends (see figure). Find the dimensions of the table, given that the area of the rectangular
portion is to be p times the sum of the areas of the circular ends, and the perimeter of the whole
table is to be 40 feet.

3. Study the solution to the mixture problem, no.3, in the previous problem set. Vary a
parameter of your choice and describe how the solution changes.

4. A 30 foot ladder and a 40 foot ladder are positioned so as to cross each other in an alley.
That is, they are leaning up against opposite walls with their bases snug up against the base of the
opposite walls. Given that they cross at a point h feet above the floor of the alley, determine the
width of the alley.

3.9 Defining your own Maple words.

This is a good place to learn how to develop and define Maple words in a worksheet. The idea is
very simple:

A. Name the given quantities –
As you are solving a problem or developing an algorithm, assign the given quantities to appro-

priate names which you have chosen. Put these assignments into a single input cell, which we will
call the parameter cell.

B. Compute the desired quantity –
Use the names assigned in your maple statements which you make in your algorithm. This part

will change and grow as you develop the definition, but try to keep all the statements in one input
cell, which we will call the definition cell. Once you get the desired output,

C. Make the Definition –
This involves mostly choosing a name for the procedure, inserting it at the top of the definition

cell in the line
name := proc(p1,p2,...,pn)
which begins the definition by assigning the name and declaring the input parameters p1, p2,

etc.
At the bottom of the definition cell is the word
end;
which signals the end of the definition.
Now, when you execute the definition cell, you should get a nicely formatted version of the

definition as output.
This is not the only set of steps you could follow when developing a definition, but it is very

natural one and works well for small definitions.

By way of example, suppose we wanted to define a word swim which would return the
solution to the swimming pool problem above.

Copy all of the input cells used to obtain the solution into one input cell.
Then all we need to do to define the word is to insert a proc line at the top and an end at the

bottom.
When inserting the proc line, you must decide what the inputs for the word will be. We will

let the ratio of the length of the pool to the width of the pool be the only input in this word. We
were calling that p in the equations, so use the same name in the proc line.

> swim := proc(p)

> eq1 := l = p*w;

> eq2 := l = w + 40;

> sol := solve({eq1,eq2},{l,w});
> end;

Warning, ‘eq1‘ is implicitly declared local

Warning, ‘eq2‘ is implicitly declared local

Warning, ‘sol‘ is implicitly declared local

swim := proc(p)

local eq1 , eq2 , sol ;

eq1 := l = p ∗ w ; eq2 := l = w + 40 ; sol := solve({eq1 , eq2}, {l, w})
end proc

When you execute the input cell containing the definition, the word swim has been added to
the vocabulary and can be used like any other Maple word. The assignments made in the definition
are declared local and a warning is issued unless you declare the names either local or global
. Usually, you will want any assignments to be local, just in case you are using the same name
outside the word to mean something else.

The output from a word is the output from the last line before the end line. So, for example,
if the pool is 3 times as long as it is wide then p = 3 and we would say

> swim(3);

{w = 20, l = 60}
On the other hand, if we put in a ridiculous input what would happen?
> swim(-4);

{w = −8, l = 32}

You can redefine the word if there is something that needs changed. For example, the swimming
pool problem doesn’t really have a solution if p is not positive, so we could insert an error trap
here.

> swim := proc(p)

> if not type(p,name) and not p > 0 then ERROR(‘oops‘)

> else eq1 := l = p*w;

> eq2 := l = w + 40;

> sol := solve({eq1,eq2},{l,w});

> fi

> end;

Warning, ‘eq1‘ is implicitly declared local

Warning, ‘eq2‘ is implicitly declared local

Warning, ‘sol‘ is implicitly declared local

swim := proc(p)

local eq1 , eq2 , sol ;

if not (type(p, name) or 0 < p) thenERROR(oops)

else eq1 := l = p ∗ w ; eq2 := l = w + 40 ; sol := solve({eq1 , eq2}, {l, w})
end if

end proc

Now redefine swim and check –

> swim(-4);

Error, (in swim) oops

Usually, one doesn’t spend a great deal of time inserting error traps in word definitionst. There
are much more interesting things to do with the mathematics of the situation. For example, suppose
we wanted to change the inputs to allow for changing the 40, the amount the length exceeds the
width, that occurs in the problem. Then simply copy down the definition into a new input cell and
make the appropriate changes. Something like this will work:

> restart;
> swim := proc (p,excess)

local eq1, eq2, sol;
eq1 := l = p*w;

> eq2 := l = w+excess;
sol := solve({eq1, eq2},{l, w})
end;

swim := proc(p, excess)

local eq1 , eq2 , sol ;

eq1 := l = p ∗ w ; eq2 := l = w + excess ; sol := solve({eq1 , eq2}, {l, w})
end proc

> swim(3,40);

{l = 60, w = 20}
> swim(3,ex);

{w =
1

2
ex , l =

3

2
ex}

Notice that you can ’read off’ the solution in words if you put in a variable for the excess. So,
if the pool is 3 times as long as it is wide, then the width must be 1/2 of the excess and the length
is 3/2 of the excess.

Visual Checking of answers.
You can also define a word to draw a picture of the pool. These types of words are very useful

to perform a visual check of computations. Sometimes, you have solved a problem incorrectly, but
do not discover that until you have performed a visual check on the solution you have obtained.

To develop a visual check of our swimming pool problem, we can first draw one picture. Get
the dimensions of a pool

> dims := swim(3,40);

dims := {l = 60, w = 20}
Now we want to draw a rectangle which is 20 by 60. We can set up a general rectangle with

one corner at the origin and the opposite corner at [l,w]
> rect := [[0,0],[l,0],[l,w],[0,w]];

rect := [[0, 0], [l, 0], [l, w], [0, w]]

> pool := subs(dims,rect);

pool := [[0, 0], [60, 0], [60, 20], [0, 20]]
> plots[polygonplot](pool,

color=blue,style=patch, scaling = constrained);

Now we can define the word. Copy down and join the appropriate input cells. Then insert a
proc line at the top (deciding on what the inputs will be), and an end line at the bottom of the
new input cell.

> drawpool := proc(p,excess)
dims := swim(p,excess);
rect := [[0,0],[l,0],[l,w],[0,w]];

> pool := subs(dims,rect);
> plots[polygonplot](pool,

color=blue,style=patch, scaling = constrained);
end;

Warning, ‘dims‘ is implicitly declared local

Warning, ‘rect‘ is implicitly declared local

Warning, ‘pool‘ is implicitly declared local

drawpool := proc(p, excess)

local dims , rect , pool ;

dims := swim(p, excess) ;

rect := [[0, 0], [l, 0], [l, w], [0, w]] ;

pool := subs(dims , rect) ;

plotspolygonplot (pool , color = blue, style = patch , scaling = constrained )

end proc
> drawpool(3,40);

Now we can make an animation of the how the pool dimensions change as we vary the excess
and/or the ratio of length to width. Make a list of the outputs from drawpool (call it movie, say)
and then use plots[display] with the option insequence = true.

> movie := [seq(drawpool(3,10*i),i=1..8)]:

> plots[display](movie,insequence=true);

This movie merely shows the dimensions of the pool increasing linearly with the excess, as we
already knew, so not much is gained from observing the animation. Often, however, one can learn
a lot from animations, if some care is taken with their construction.

3.10 Problems

Exercise: Make a movie of the change in the pool as the ratio p changes from 1 to 4 by increments
of 1/2.

Exercise: What is wrong with taking the original word swim and changing to proc line to read
swim := proc(p,w) ?

Exercise: Modify drawpool so that if the ratio p is greater than 2, the color of the pool is blue,
otherwise the color is red.

Exercise: Define a word quadform to take three numbers a, b, c and return the roots of the
quadratic equation a x2 + b x + c = 0, if they are real, and the message ’no real roots’ if they are
complex. Don’t just use solve here .

Exercise: Modify the word swim so that it simply returns the point [l,w] rather than the solution
to the equations.

Exercise: Define a word which takes a function and tabulates its value over a given interval
with a given increment between x values.

4 More worked Problems

In this chapter, we have gathered some more problems, their solutions, and followup questions for
your inspection. None of these problems require any calculus for their solution. Perhaps you can
get some ideas from these.

4.1 A billiard ball problem.

A cue ball is 2 feet from the edge of the table. The player wants to bounce the cue ball off the
edge and hit the red ball which is 3 feet from the same edge. If the distance between the vertical
projections of the cue ball and the red ball to the edge is 5 feet, find the angle at which the player
should bounce the ball off the edge.

A Solution.
First draw a picture and label the unknowns. We can think of the edge as the x-axis and the

cue and red balls being located at (0,2) and (5,3) respectively. Then let (x,0) be the point on the
x-axis where the ball should bounce. The angle (0,2), (x,0), (0,0) of incidence is going to be the
same as the angle (5,0),(x,0),(5,3) of reflection.

We can draw several ’possible’ paths, but only one will satisfy the incidence = reflection property.

> cue := [0,2]: red := [5,3]:

> path := x -> plot([cue,[x,0],red]) ;

path := x → plot([cue, [x, 0], red ])

> plots[display]([seq(path(i),i=0..5)],scaling=constrained);

We can express the tangent of the angle in two ways. All we need to do is solve the resulting
system of equations for a and x.

> eq1 := tan(a)=2/x;

eq1 := tan(a) =
2

x
> eq2 := tan(a)=3/(5-x);

eq2 := tan(a) =
3

5 − x
> sol := solve({eq1,eq2},{a,x});

sol := {x = 2, a =
1

4
π}

So the player should bounce the ball off at a 45 degree angle. This is the way most contrived
problems work out. If we were to jiggle the distance of the cue ball from the edge a little, we
would expect the angle to jiggle too. That is where parameterization comes into play. We could
parameterize this problem in several ways. For example, replace the distance of the cue ball from
the edge with a parameter h and rework the problem.

> eq1 := tan(a)=h/x;

eq1 := tan(a) =
h

x
> eq2 := tan(a)=3/(5-x);

eq2 := tan(a) =
3

5 − x
> sol := solve({eq1,eq2},{a,x});

sol := {x = 5
h

h + 3
, a = arctan(

1

5
h +

3

5
)}

Now we can investigate questions such as:
Question: What happens to the angle as the cue ball moves away from the edge? towards the

edge?
A Solution: The ball moves away from the edge as h gets larger. By looking at the expression

for a in terms for h, we see that the angle gets closer to 90 degrees. On the other hand as h gets
closer to 0 (as the ball moves towards the edge) the angle gets closer to arctan(3/5), or

> evalf(180/Pi*arctan(3/5));

30.96375653

degrees.
Exercise: Define a word drawbounce := proc(h) which takes h, the distance of the cue ball from

the edge and returns a picture of the cue ball, the red ball and the path the cue ball takes if it
bounces once off the lower edge of the table. Leave all the other data the same as given in the
original problem.

Exercise: Use the word drawbounce to make a movie of what happens to the path of the cue
ball as the cueball starts from a point closer to the edge of the table.

4.2 A Variation on the Billiard Ball Problem.

Exercise: Suppose that the edge of the billiard table opposite the x-axis is 4 feet wide and the
player wants to bounce the cue ball twice before it hits the red ball. What angle should he bounce
it off the edge?

Solution.
The second edge has equation y = 4 , and the second bounce point will be (x2,4). Notice that

the angle of reflection of the first bounce equals the angle of incidence of the second bounce, since
the two edges are parallel. So we have three ways two express the angle a in terms of x and x2.

> eq1 := tan(a)=2/x;

eq1 := tan(a) =
2

x
> eq2 := tan(a)=1/(5-x2);

eq2 := tan(a) =
1

5 − x2
> eq3 := tan(a)=4/(x2-x);

eq3 := tan(a) =
4

x2 − x

Now solve the equations for x,x2, and a.
> solve({eq1,eq2,eq3},{x,x2,a});

{a = arctan(
7

5
), x =

10

7
, x2 =

30

7
}

> evalf(convert(arctan(7/5),degrees));

54.46232221 degrees

So the player should bounce the ball off at about a 54.5 degree angle. Comparing this with the
first problem, we note that the bounce angle should be larger than 45 degrees since the player must
bounce the ball twice.

Exercise: What is the angle if the cue bounces off the edge
y = 4

first?

4.3 Water tank problem.

Problem: Suppose you are standing at the point (0,0,4) on the z-axis, looking down towards the
x-axis. A spherical water tank 2 units in diameter is centered on the origin (0,0,0). You can’t ’see’
the points on the x-axis close to (0,0,0) because the water tank is in the way, but you can see points
far out on the x-axis. So, for example, the line connecting the point (10,0,0) to (0,0,4) doesn’t
touch the water tank (does it?), so you can ’see’ that point. The question is, what is the smallest
point (a,0,0) on the positive x-axis you can see from (0,0,4).

Solution.
Let (c,0,d) be the point on the tank such that the line through (0,0,4) and (c,0,d) is tangent

to the tank. So the triangle with vertices (0,0,0), (c,0,d), and (0,0,4) is a right triangle. Using
Pythagoras’ theorem, this means that

> eq1 := c^2 + d^2 + c^2 + (d-4)^2 = 4^2;

eq1 := 2 c2 + d2 + (d − 4)2 = 16

But also

> eq2 := c^2+d^2=1;

eq2 := c2 + d2 = 1

Since the three points [a,0,0], [c,0,d], and [0,0,4] are collinear,
> eq3 := (d-0)/(c-a)=4/(-a);

eq3 :=
d

c − a
= −4

a
> solve({eq1,eq2,eq3},{c,d,a});

{a = 4RootOf(15 Z 2 − 1), d =
1

4
, c =

15

4
RootOf(15 Z 2 − 1)}

By inspection, we see that a = 4√
15

is the smallest positive real number the observer can see

from (0,4).
We can draw a picture of the observer, the water tank, the line of sight from the observer to

the x-axis.
> tank := plot3d(1,theta=0..2*Pi,phi=0..Pi/2,coords=spherical):
> lineofsight := plots[polygonplot3d]([[0,0,4],[4/sqrt(15),0,0]],

thickness=2,color=black):
> plots[display]([tank,lineofsight],

scaling=constrained,axes=boxed);

Exercise: Suppose you are at the point (0,0,h) on the z-axis looking toward the positive
x-axis. What is the smallest point you can see on the positive x-axis? Where are you if the smallest
point you can see is (10,0,0)?

Question: What happens if we replace the spherical tank with a parabolic cylinder, say
z = 1 − x2?

4.4 A ladder problem.

Problem: The foot of a ladder is placed at [3,0] and leaned up against a parabolic hill y = 3 − x2

. Where does the ladder cross the y-axis?
Solution:
Draw a picture, showing the hill and the ladder. Mark the point where the ladder crosses the

y-axis [0,a]. Mark the point where the ladder touches the hill, [b, 3 − b2]. The ladder is tangent to
the hill at [b, 3 − b2]. We could figure out a if we knew the slope of the ladder. So take a point close
to ( b, 3 − b2) on the hill. Say the point is ( b + h, 3 − (b + h)2) where h is a small number (but
not zero). The approximate slope of the ladder is the slope of the line through these two points.

Calculate this slope, simplifying as much as possible. Then find the limiting value of this slope as
h approaches 0, that is, as the point [b + h, 3 − (b + h)2] moves toward [b, 3 − b2] on the parabola.
To find the slope using Maple,

> f := x -> 3 - x^2;

f := x → 3 − x2

> approxslope := (f(b+h) - f(b)) / ((b+h) - b);

approxslope :=
−(b + h)2 + b2

h
> approxslope := simplify(approxslope);

approxslope := −2 b − h

> slope1 := limit(approxslope,h=0);

slope1 := −2 b

But the slope of the ladder is also the slope of the line through [3,0] and [b, 3 − b2] .
> slope2 := (3-b^2)/(b-3);

slope2 :=
3 − b2

b − 3

Now set these slopes equal and solve for b.
> sol := solve(slope1=slope2,b) ;

sol := 3 +
√

6, 3 −
√

6

Which solution do we want? It will have to be the smaller term (why?) After having decided
that, find a by finding where the ladder crosses the y-axis.

> b := min(sol);

b := 3 −
√

6

> ladder := x -> slope2*(x-b) + f(b);

ladder := x → slope2 (x − b) + f(b)

> a:= ladder(0);

a := −1

6
(3 − (3 −

√
6)2)

√
6 (−3 +

√
6) + 3 − (3 −

√
6)2

> simplify(a);

18 − 6
√

6

> evalf(a,4);

3.303

> plot({f,ladder},0..3,color=black);

4.5 Another Ladder Problem.

Problem: A 5 foot ladder is leaning up against a parabolic hill y = 3 − x2. The foot of the ladder
is on the positive x-axis, and the top of the ladder is on the y-axis. Describe the position of the
ladder, by locating the coordinates of the foot of the ladder and the top of the ladder. Note: This
problem may have two different solutions. Draw a picture and show how two positions may be
possible.

Solution:
First define the hill. The ladder leans up against the hill at the point [b,f(b)], and will be

tangent to the hill there. By our calculations in the previous problem, the slope of the ladder will
be −2 b at that point.

> restart;

> f := x->3-x^2;

f := x → 3 − x2

> slope := -2*b;

slope := −2 b

So we could define the tangent line function at (b,f(b)). That will be the ladder.
> tang := x -> -2*b*(x-b)+ f(b);

tang := x → −2 b (x − b) + f(b)

Then get the x and y intercepts of the ladder, call them xb and yb.

> xb := solve(tang(x)=0,x);

xb :=
1

2

b2 + 3

b
> yb := tang(0);

yb := b2 + 3

The ladder is 5 feet long. This gives us the equation we need to determine the value of b.

> eq1 := xb^2+yb^2=5^2;

eq1 :=
1

4

(b2 + 3)2

b2
+ (b2 + 3)2 = 25

Let’s manipulate this equation.

> eq2 := expand(4*b^2*simplify(eq1));

eq2 := 25 b4 + 4 b6 + 42 b2 + 9 = 100 b2

> eq3 := lhs(eq2)-rhs(eq2);

eq3 := 25 b4 + 4 b6 − 58 b2 + 9

Now we are interested in finding the roots of eq3, where the graph of eq3 crosses the b axis. So
plot the graph. First we graphed it over the interval 0..3, but then narrowed it down to the interval
0..1.7.

> plot(eq3,b=0..1.7);

Indeed, we can see by looking that there are two positions for the ladder, a high position and
a low position.

> sol := fsolve(eq3,b,0..1.5);

sol := 0.4093966141, 1.289029267

Now, to draw a picture, showing the hill and both positions of the ladder.
> plot({f,subs(b=sol[1],op(tang)),
> subs(b=sol[2],op(tang))},0..4,0..5,color=black);

0

1

2

3

4

5

1 2 3 4

4.6 Variation on the last ladder problem.

Exercise: In the ladder problem above, suppose we shorten the ladder. We can imagine that as
the ladder gets shorter, the two positions the ladder can occupy get closer to each other, until at
last they coincide. What is the shortest ladder that can lean against the hill so that its foot is on
the x-axis and its top is on the y-axis?

Solution:
Go back to the ladder and look at the expression xb2 + yb2 . This represents the square of the

length of the ladder. What we can do is plot the square root of this expression over the interval
0..3 and find the low point on the graph.

> plot(sqrt(xb^2+yb^2),b=0..3,y=0..20);

The shortest ladder is around 4.3 feet long.

48



5 Differentiation and its uses.

In the previous chapter, we looked at some tangent line problems which we solved by finding the
limit of a ratio. For the function f(x) = x2 + 10, we found that the slope of the tangent line to the
graph of f at a point [b,f(b)] is 2b. The defines a function, called the derivative of f, d

dx f(x) = 2x.
It turns out that the derivative function also measures the rate of change of f. The notion of rate of
change has application to almost any area of knowledge. In this chapter, we begin to make serious
use of differentiation in solving problems which involve rate of change.

5.1 Defining Derivatives

The Maple word limit can be used to calculate derivatives from the definition. Say f(x) = x2 + x − 3
and you want to compute f ’ by the definition. So enter

> f := x -> x^2 + x -3;

f := x → x2 + x − 3

> fp := limit((f(x+h)-f(x))/h,h=0);

fp := 2x + 1

You would get the same thing by entering

> fp := diff(f(x),x);

fp := 2x + 1

The main Maple words here are diff and limit but in practice you will use diff much more.
Derivatives of functions defined by expressions, such as

> y := x^2 + x;

y := x2 + x

would be computed with, as follows:
> yp := diff(y,x);

yp := 2x + 1

Now evaluate the derivative of y at x = 3 using subs,

> subs(x=3,yp);

7

to get the derivative at x = 3 . Another way to do this is with unapply , a Maple word to
change an expression into a function.

> yp:=unapply(yp,x);

yp := x → 2x + 1

> yp(3);

7

Yet another way is to use the D operator , which takes a function and returns its derivative
as a function.

> f := x -> x^2 + x;

f := x → x2 + x

> df := D(f);

df := x → 2x + 1

> df(3);

7

5.2 The student package

Maple contains a package of words, called the student package , that pertain to the study of
calculus. This is a good spot to examine some of them. In order to load the words for the student
package, enter

> with(student);

[D, Diff , Doubleint , Int , Limit , Lineint , Product , Sum, Tripleint , changevar ,

completesquare , distance , equate, integrand , intercept , intparts , leftbox , leftsum,

makeproc, middlebox , middlesum , midpoint , powsubs , rightbox , rightsum ,

showtangent , simpson , slope , summand , trapezoid ]

One interesting word is showtangent . Give it a function and a point where the derivative of
the function is defined. The word showtangent will return a picture of the function and the tangent
line to the function at the given point. So to see the graph of y = x3 + x − 1 and its tangent line
at x = 4.5 , enter

> showtangent(x^3+x-1,x=4.5,color=black);

5.3 Problems

Exercise: Use each word unapply and D to define the derivative of ? Plot both f and its derivative
over the interval [-1..1]

> f := x->x^4 - 2*x^3 + x -3 ;

f := x → x4 − 2x3 + x − 3

> df := D(f);

df := x → 4x3 − 6x2 + 1

> plot({f,df},-2..2);

Exercise: Use ’for x from -1 by .25 to 2 do’ to tabulate the values of f and its derivative at
increment of .25 over the interval from -1 to 2.

3. Plot f and its derivative over this interval. Then use showtangent to plot f and the tangent
line to the graph at x = −5.

4. Where does f cross the x-axis? Find the zeros of f ’(x). That is, find the extreme values of f.

5.4 Newton’s Method.

You have used fsolve to solve an equation f(x) = 0 for x. How does that work? It possibly might use
Newton’s method, which is a very fast method for solving such equations when f is differentiable.
The idea is simple. By graphing or guessing, you come up with a first estimate of the solution, call
it x0. Now the point (x0,f(x0)) is probably not on the x-axis. Iif it were then f(x0) is 0 and x0 is
a solution. So we go along the tangent line to the graph of f until we come to the x-axis, at the
point (x1,0) say. We find that we can express x1 in terms of x0, f(x0), and f’(x0). This number x1
is the second approximation to the solution.

> restart;

> eq := diff(f(x0),x0) = (f(x0)-0)/(x0-x1);

eq := d
dx0 f(x0 ) =

f(x0 )

x0 − x1
> expand(solve(eq,{x1}));

{

x1 = x0 − f(x0 )
d

dx0 f(x0 )

}

This equation is called Newton’s method . It has the property that you can iterate it, that is,
after computing x1 you rename it x0 and apply the method again. You can continue this iteration

until sequence of numbers converges, or it becomes clear that it won’t converge. Here is a version
of Newton’s method which draws a so-called stairstep picture to go with the calculations The final
approximation is the title.

> vnewt := proc(f,start,a,b,iterations)
local i, x0, fp, p,pl1,pl2;
x0 := evalf(start);

> fp := D(f);
p := [x0,0];
for i from 1 to iterations do

> p := p, [x0, f(x0)];
x0 := x0 - f(x0)/fp(x0);
p := p,[x0,0];

> od;
pl1 := plot( [p] ,x=a..b, color=black);
pl2 := plot(f,a..b,color=red) ;

> ### WARNING: semantics of type ‘string‘ have changed
plots[display]([pl1,pl2],title=convert(x0,string));
end:

Now test vnewt to solve x2 = 2 by starting starting at 3 and using a plot interval of 0 to 4..
> vnewt(x -> x^2 - 2, 10 , 0, 10 ,5);

1.414525655

Exercises.
Exercise: Use vnewt to solve cos(x) − x = 0 for x for various starting points from 2 to 3.05.

Are there starting points which don’t lead to a solution? Explain your answer.

Exercise: Use vnewt to solve x2 + .1e-1 = 0 . Try start at .2, a at -.3, b at .3. What happens?

Exercise: Define a new version of vnewt, tnewt, which has the same inputs, but returns the
list of approximations to the root. Test your word.

5.5 Use of the derivatives in plotting

There is a close connection between the first and second derivative of a (differentiable) function f
and what its graph looks like. The local extreme points of f occur amongst the critical points
(the zeros of f ’ ), and the inflection points occur amongst the zeros of f ”. Also f is increasing
(decreasing) on intervals where f ’ is positive (negative), and f is concave up (down) on intervals
where f ” is positive (negative).

For example, let’s look at the function

> y := (x^5-1)/(6*x^2+1);

y :=
x5 − 1

6x2 + 1

First plot it over an interval, say from −2 to 2.

> plot(y,x=-2..2);

Look at the graph carefully and estimate the coordinates of local extrema and inflection
points . Now lets get these points exactly using some calculus. First compute the derivative of y
with respect to x.

> yp := diff(y,x);

yp :=
5x4

6x2 + 1
− 12 (x5 − 1)x

(6x2 + 1)2

> yp := simplify(yp);

yp :=
x (18x5 + 5x3 + 12)

(6x2 + 1)2

Get the critical points by solving yp = 0 . Actually, we get better results by finding when the
numerator of yp is 0.

> cp := fsolve(numer(yp),x);

cp := −0.8657650194, 0.

cp is an expression sequence of two critical points, cp[1] and cp[2]. By inspecting the graph, we
can see that cp[1] is a local maximum and cp[2] is a local min. Let’s evaluate y at x = cp 1, and at
x = cp2.

> subs(x = cp[1],y);

−0.2703889018

> subs(x = cp[2],y);

−1.000000000

Now we know the coordinates of the extreme points on the graph. Next let’s get the coordinates
of the inflection points.

> ypp := diff(yp,x);

ypp :=
18x5 + 5x3 + 12

(6x2 + 1)2
+

x (90x4 + 15x2)

(6x2 + 1)2
− 24x2 (18x5 + 5x3 + 12)

(6x2 + 1)3

> simplify(ypp);

4 (54x7 + 27x5 + 5x3 − 54x2 + 3)

(6x2 + 1)3

> pip := fsolve(numer(ypp),x);

pip := −0.2324164053, 0.2392938071, 0.8746375912

> for i from 1 to 3 do print(pip[i], subs(x=pip[i],y)) od;

−0.2324164053, −0.7557396749

0.2392938071, −0.7437022342

0.8746375912, −0.08732685718

So we see there are three inflection points.
Exercise: there always an inflection point between two extreme points? Explain your answer.
Exercise: Is there always an extreme point between two inflection points? Explain your

answer.

5.6 Implicit Differentiation

Suppose you have a curve in the plane which is defined by an equation in x and , say f(x, y) = 0.
If [x0,y0] is a point on the curve, that is f(x0 , y0 ) = 0, then we can ask for an equation for the
tangent line to the curve at that point. One way to obtain the slope of the tangent is as follows:
assume that y is a function of x near [x0,y0]. Then near [x0,y0], f(x,y(x)) is the constant function 0,
so when we differentiate f(x,y(x)) with respect to x the result function of x is the constant function
0. But also, the function we get will be linear in y’(x). That means we can solve for y’(x) in terms
of x and y(x). Evaluating this at [x0,y0] gives the slope of the tangent. This algorithm is referred
to as implicit differentiation .

Exercise: Plot the equation x2 sin(x y) + y3 − 5 = 0 in the rectangle x=0..4, y= 0..2, and draw
a tangent to the graph at the point [3,y0] where y0 is between 0 and .6.

Solution:
First we will draw the curve using plots[implicitplot].

54



> plots[implicitplot](x^2*sin(x*y)+y^3-5 = 0,x=0..4,y=0..2);

Now we need to find the desired y0. Use fsolve for that.

Error, (in plot/iplot2d/expression) bad range arguments x = 0 .. 4,
(x^5-1)/(6*x^2+1) = 0 .. 2

> y0 := fsolve(3^2*sin(3*y)+y^3-5=0,y,0..(.6));

Error, (in fsolve) fsolve cannot solve on (x^5-1)/(6*x^2+1)

Now let’s get a formula for y’(x). First substitute y(x) in for y.
> g := subs(y=y(x),x^2*sin(x*y)+y^3-5=0) ;

g := x2 sin(
x (x5 − 1)

6x2 + 1
) +

(x5 − 1)3

(6x2 + 1)3
− 5 = 0

Then differentiate the resulting function of x with respect to x.
> dg := diff(g,x);

dg := 2x sin(
x (x5 − 1)

6x2 + 1
) + x2 cos(

x (x5 − 1)

6x2 + 1
) (

x5 − 1

6x2 + 1
+

5x5

6x2 + 1
− 12x2 (x5 − 1)

(6x2 + 1)2
)

+
15 (x5 − 1)2 x4

(6x2 + 1)3
− 36 (x5 − 1)3 x

(6x2 + 1)4
= 0

Notice the left hand side is linear in y’(x), as we asserted it would be. So solve the equation for
y’(x).

> yp := solve(dg,diff(y(x),x));

yp :=

Now turn this back into an expression in x and y.
> yp := subs(y(x)=y,yp);

yp :=
x(x)5 − 1

6 x(x)2 + 1
=

x5 − 1

6x2 + 1
Calculate the slope of the tangent at [3,y0].
> slope :=evalf(subs({x=3,y=y0},yp));

slope :=
3(3)5 − 1.

6. 3(3)2 + 1.
= y0

Write the equation for the tangent line in point slope form.
> tangent := y - y0 = slope*(x-3);

tangent :=
x5 − 1

6x2 + 1
− y0 = (4.400000000x − 13.20000000 = (x − 3) y0 )

Plot the curve and tangent separately so that we can color them differently. Then display them.
> curve :=plots[implicitplot]( x^2*sin(x*y)+y^3-5 = 0

,x=0..4,y=0..2,thickness=2):

Error, (in plot/iplot2d/expression) bad range arguments x = 0 .. 4,
(x^5-1)/(6*x^2+1) = 0 .. 2

> tanline := plots[implicitplot]( tangent ,x=0..4,y=0..2,thickness =
2, color=blue):

Error, (in simpl/relopsum) invalid terms in sum

> plots[display]([curve,tanline]);

Exercise: Find a point on the curve in the rectangle where the tangent line is vertical. (Hint:
you will need to assume x is a function of y and differentiate with respect to y here.)

Error, (in plots/display) first argument must be a plot structure or
list, set or array of plot structures

5.7 Max-min Problems

A vast number of problems fall into the category of the so-called max-min problems. These are
problems in which some quantity, Q, is to be maximized or minimized as needed. The quantity Q
is a function of one or more other variables which are subject to some constraints. We attempt to
use these constraints to eliminate all but one of the variables in the computation of Q. If we are
successful in doing this, we can then use calculus to locate the maximum or minimum if it exists.

Let’s take an example.

5.7.1 A Paper folding problem:

Problem: A sheet of paper 4 inches wide by 8 inches high is folded so that the bottom right corner
of the sheet touches the left hand edge of the sheet. The tip of the corner is no more than 4 inches
above the bottom edge of the paper. Then the paper is creased (see figure). Find the length L of
the crease, and find how to fold the paper so that L is minimum.

Solution:
Let h, x, and y be as shown in the diagram below.
> restart;
> A1:=plots[textplot](

{[2.6,1.9,"L"],[3.7,2,"h"],[.1,1.1,"y"],[.5,.3,"x"]},align=RIGHT):
> A3:=plot({[[0,0],[0,8],[4,8],[4,4.45],[1.219,0],[0,0]],[[4,4.45],[0,

2.5],[1.219,0]]},color=blue):
> A4:=plot([[4,4.45],[4,0],[1.219,0]],color=red):
> A5:=plots[polygonplot]([[4,4.45],[0,4.45],[0,2.5]],style=patch,color=

tan):

> plots[display]([A1,A3,A4,A5],axes=boxed,scaling=constrained);

We can see several equations relating x, y, h, and L in the diagram. For example, the small
right triangle with legs x and y has a hypotenuse which is 4-x units long. This gives eq1.

> eq1 := y^2+x^2=(4-x)^2;

eq1 := y2 + x2 = (4 − x)2

eq2 comes from the right triangle with hypotenuse L and legs h and 4-x.

> eq2 := L^2 = (4-x)^2 + h^2;

eq2 := L2 = (4 − x)2 + h2

Now it is easy to work out that the tan right triangle with hypotenuse h and legs 4 and h-y is
similar to the right triangle with hypotenuse 4-x and corresponding legs y and x. So we get eq3.

> eq3 := 4/(h-y)=y/x;

eq3 :=
4

h − y
=

y

x

> h := solve(eq3,h);

h :=
4x + y2

y
> x := solve(eq1,x);

x := −y2

8
+ 2

> L := unapply(sqrt(op(2,simplify(eq2))),y);

L := y → 1

8

√

(16 + y2)3

y2

So we have L expressed as a function of one variable y. Examining the behavior of L as y varies,
> plot(L,2..4);

we see that there is a minimum length crease of about L = 5.2 inches at about y = 2.8 inches.
We can get a more precise value with fsolve, by locating the x between 2.6 and 3 where the derivative
of the crease function is 0.

> y1 := fsolve(diff(L(y),y),y,2.6..3);

y1 := 2.828427125

Now check the value of x and L for this value of y.

> minL := L(y1);

minL := 5.196152422

> minx := subs(y=y1,x);

minx := 1.000000000

Notice the nice integer value of minx . This gives rise to a simple construction of the crease of
minimum length: by folding twice along the bottom edge, we can mark the point 1 inch from the
left edge of the paper. Then bring the corner point up to the left edge and crease.

Exercise: Use Maple to draw the diagram showing the minimum length crease.

Exercise: Suppose we wanted to minimize L + y, rather than just L. Would the minimum occur
at the same place? Work it out.

6 More Max-min Problems

6.1 Stumbling onto max-min Problems

Sometimes, in the process of working on one problem, which may not be a max-min problem
at all, one may stumble onto an interesting max-min problem. This can occur naturally in a
Maple worksheet. Just take a problem in which you need to calculate a quantity in terms of some
given quantities. Then make one of the given quantities a variable quantity and go through the
calculation. At the end, you have obtained a formula expressing the desired quantity as a function
of the variable quantity. Then you can study the behavior of this function with plot and diff,
locating any extreme values it may possess. Here is an example.

A Ladder Problem: A 20 foot ladder is leaning against the top corner of a bay window in
such a way that the top of the ladder is just resting against the wall of the house, and the base
of the ladder is on the ground. The top corner of the window is 3 feet out from the house and 12
above the ground. How high up on the wall is the top of the ladder touching?

> restart;

> with(plots):

> A1:=plot({[[0,15],[0,-1]],[[0,12],[3,12],[3,4],[0,4]]},color=navy):
> A2:=plot({[[-1,0],[16,0]],[[0,14.937],[15.257,0]]},color=black):
> A3:=textplot({[-.5,13,’h’],[8,8,’d’],[-.75,6,‘12‘],[1.5,11.25,‘3‘]

}):
> display({A1,A2,A3},axes=none,scaling=constrained,style=line);

d

h

3

12

Solution:
Let’s parameterize this problem by changing the length of the ladder to a parameter, say d. If

we set up a coordinate system so that the wall is the positive y-axis, the ground is the positive
x-axis, and the corner of the bay window is at (3,12), then by similar triangles, h/sqrt(hˆ2+3ˆ2) =
(h+12)/d , where h+12 is the height above the ground the ladder touches the wall. Let’s let Maple
do some work. First set up the equation and simplify it.

> restart;

> eq := h^2/(h^2+9) - (h+12)^2/d^2;

eq :=
h2

h2 + 9
− (h + 12)2

d2

Let’s manipulate eq a little.

> eq := (collect(numer(normal(eq)),h));

eq := −h4 − 24h3 + (−153 + d2)h2 − 216h − 1296

If we solve for the height above the wall, h, in terms of the length of the ladder, d, we get more
extraneous solutions than if we solve eq for d in terms of h.

> sol := [solve(eq,d)];

sol := [

√
h2 + 9 (h + 12)

h
, −

√
h2 + 9 (h + 12)

h
]

Clearly, we are interested in the positive values for d, ie, sol 1.

> plot(sol[1],h=3..7);

Upon examination of the graph, we see that there are two solutions to the original problem. In
fact, and this is clear if we reflect on the situation, there will generally be two positions the ladder
can occupy if it is long enough, no positions if it is too short, and exactly one position if the ladder
exactly the right length. That length is slightly under 20 feet, the y-coordinate of the base of the
graph. We can discover this length easily.

> hmin := fsolve(diff(sol[1],h),h,4..5);

hmin := 4.762203156

This gives the height above the bay window that the ladder touches.

> dmin := subs(h=hmin,sol[1]);

dmin := 19.81098308

So, we have stumbled onto the solution of the problem of finding the shortest ladder such that
there is a solution to the problem.

6.2 Problems:

Wandering Critical Points: For the cubic f(x) = x3 + a x2 − x + 3, plot in the same window
the graphs of f for several positive values of a. What happens to the local extreme points and the
inflection points of f as a increases from 0 towards infinity? Also, try to use the word animate to
do the same thing.

> plots[animate](x^3+a*x^2-x+3,x=-12..9,a=0..10,color=blue);

Asymptotes
Make a sketch of the function y = x3−8

x3+2 x2 . Find all asymptotes of the function. Also, find the
intercepts, the local maxima, local minima, and inflection points of the function.

Two Pigpen problems
Pigpen I: A pigpen (shed for holding pigs) is to be shaped like a box with no floor and no front.

The top is to be square, and the volume is to be 18 cubic yards. What dimensions should the
pigpen have in order to minimize the amount of material used to construct it?

Pigpen II: Instead of requiring the top to be square, suppose we want the top to be b times
as wide as it is long, where b is some fixed positive number. Find the minimum amount, A, of
material needed to construct such a pigpen as a function of b. Draw a conclusion about the most
economical shape for the top of the pigpen.

Big Clock Problems
Big Clock I: A tower clock has a 4 foot minute hand and a 3 foot hour hand. How fast are the

tips of the hands moving apart at 12:15?

Big Clock II: At what time are the hands moving apart most quickly?

6.3 Solutions

Wandering points
Define the function and compute its derivatives.

> restart;

> f:= x -> x^3 + a*x^2 - x + 3;

f := x → x3 + a x2 − x + 3

> fp := unapply(diff(f(x),x),x);

fp := x → 3x2 + 2 a x − 1

> fpp := unapply(diff(f(x),x,x),x);

fpp := x → 6x + 2 a

Let’s plot f over x from -6 to 4 and a from 0 to 5.
> plot({seq(f(x),a=0..5)},x=-6..4); a := ’a’:

–200

–150

–100

–50

50

100

150

–6 –4 –2 2 4x

Looking at the graphs suggests that the local maximum (the leftmost critical point) wanders
off to the left and drifts upward. The other critical point and the inflection point seem to stay
stationary, although that may be an illusion. Let’s calculate the points and look at their limits as
a goes to infinity. First, find the x-coordinates of the critical points and the inflection point.

> cp := solve(fp(x),x);

cp := −1

3
a +

1

3

√
a2 + 3, −1

3
a − 1

3

√
a2 + 3

> ip := solve(fpp(x),x);

ip := −1

3
a

We can see that as a gets large, cp1and cp2 go to 0 and −∞, respectively.

> limit(cp[1],a=infinity);

0

> limit(cp[2],a=infinity);

−∞

The local minimum of f(x) as a approaches ∞ is (0,3), and the local maximum and inflection
points head towards (
−∞, −∞

), with the inflection point roughly halfway between the two extrema.

To observe the graph of f(x) as increases (for example, from a = 0 to a = 10 ) is a great problem
for using the word animate. Here is a sample of what you might have done. Once you have activated
the animate plot press your pointer on the ’once’ button to make the animate plot ’loop’.

> plots[animate](x^3+a*x^2-x+3,x=-12..9,a=0..10,color=blue);

–1500

–1000

–500

500

1000

1500

–10 –5 5
x

Pig Pen II
First draw a picture
> A1:=plot({[[0,0],[0,2],[2,3],[5,3],[5,1],[3,0],

[3,2],[0,2]],[[0,0],[3,0],[3,2],[5,3]]},color=navy):
> A2:=plots[textplot]({[1.5,-.25,’w’],[4.25,.35,’l’],

[5.25,2,’h’]},color=black):
> A3:=plots[textplot]([2,3.25,‘No front, no floor‘],color=black):

> display({A1,A2,A3},axes=none,scaling=constrained);

No front, no floor

w

h

l

We are given that the length I of the pen is bw, where b is a fixed positive number. The problem
is to minimize the area, A = 2 hl + hw + lw , of the pen subject to the constraint that the volume,
V=hlw , of the pen is 18 cubic feet.

> restart;

> A := 2*h*l + h*w + l*w;

A := 2h l + hw + l w

> l := b*w;

l := bw

Now the area is expressed as a function of the height and width of the pen. But the height
of the pen is h = 18/(lw) from the constraint equation. Let’s substitute for h and turn A into a
function.

> A := unapply(subs(h=18/(l*w),A),(b,w));

A := (b, w) → 36

w
+

18

bw
+ bw2

Plotting the area function for various values of b
> plot({A(1,w),A(1/2,w),A(1/10,w)},w=1..15);

200

2 4 6 8 10 12 14
w

> plot(A(1/2,w),w=1..15);

40

60

80

100

2 4 6 8 10 12 14
w

’might’ lead one to suspect that the minimum of the minimums Ab occurs at b = 1
2 . We can

express Ab as a function of b. First, find the width wb of the pen with minimum area. That will
occur at the real root of
∂

∂w A(b, w) = 0
. Then substitute that in for w in A.

> Aw := normal(diff(A(b,w),w));

> w[b] := ((9+18*b)/b^2)^(1/3);

> A[b] := unapply(A(b,w[b]),b);

> fsolve(diff(A[b](b),b),b,0..1);

Aw := 2
−18 b − 9 + b2 w3

bw2

wb := (
9 + 18 b

b2
)(1/3)

Ab := b → 36

(
9 + 18 b

b2
)(1/3)

+
18

b (
9 + 18 b

b2
)(1/3)

+ b (
9 + 18 b

b2
)(2/3)

0.5000000000

Indeed, our suspicion pans out.
Big Clock II.

By the law of cosines, the distance s between the tips of the minute hand and the hour hand is
> restart;

> s := sqrt(3^2+4^2 - 2*3*4*cos(theta));

s :=
√

25 − 24 cos(θ)

where θ is the angle between the hands. This angle is a linear function of time t measured in
minutes. At t = 0 the angle is 0 radians and after 12 (60)

11 minutes the angle is 2π radians, so in
general,

> theta := (11/12)*(2*Pi/60)*t;

θ :=
11

360
π t

To find when the distance between the minute hand and hour hand is increasing most rapidly,
lets think through what happens. At 12:00, the hands are together, but the minute hand is moving
away from the hour hand. Actually they are both moving to the right, but the minute hand moves
12 times faster. The time at which the distance between the two hands are increasing most rapidly
would be when the second derivative of s with respect to time is zero. This will occur sometime
before the hands are pointing in opposite directions, which works out to be 1

2( 12
11

)(60)
minutes past

12.

> sp := diff(s,t);

sp :=
11

30

sin(
11

360
π t)π

√

25 − 24 cos(
11

360
π t)

> spp := normal(diff(sp,t));

spp := − 121

10800

π2 (12 sin(
11

360
π t)2 − 25 cos(

11

360
π t) + 24 cos(

11

360
π t)2)

(25 − 24 cos(
11

360
π t))(3/2)

> ti := fsolve(spp,t,0..30*12/11);

ti := 7.529022202

> evalf(subs(t=ti,sp));

0.2879793266

> evalf(subs(t=ti,s));

2.645751311

So at about 7.529 minutes after 12, the distance between the tips of the minute hand and the
hour hand is increasing at the maximum rate of about .288 feet per minute. Also, the tips are
about 2.65 feet apart at that time.

7 Early Integration.

At least as many problems can be solved using integration as can be solved using differentiation,
so we need to find out about it too.

Integration involves adding up, which Maple does with sum.

7.1 Learning to use the Maple words Sum and sum .

Sum is the passive form of sum, so you can set up an equation displaying and computing the sum
of the first 100 positive integers.

> Sum(i, i=1..100)=sum(i, i=1..100);

100
∑

i=1

i = 5050

Suppose we wanted to compute the sum of the arithmetic sequence

4 + 7 + 10 + 13 + .... + 301

First thing we need to figure out is the number of terms in this sum. Well, the common difference
is 3 and the last term can be written as 4 + 3 i where

> i = (301-4)/3;

i = 99

So, there are 100 terms and we can sum this as
> Sum(4+i*3,i=0..99)=sum(4+i*3,i=0..99);

99
∑

i=0

(4 + 3 i) = 15250

Exercise: Find the sum of the square roots of i, as i goes from 1 to 10.
If we proceed as before,
> Sum(sqrt(i),i=1..10)=sum(sqrt(i),i=1..10);

10
∑

i=1

√
i = 1 +

√
2 +

√
3 +

√
4 +

√
5 +

√
6 +

√
7 +

√
8 +

√
9 +

√
10

the sum is not computed. Use evalf to convert it to a decimal.
> Sum(sqrt(i),i=1..10)=evalf(sum(sqrt(i),i=1..10));

10
∑

i=1

√
i = 22.46827818

Exercise: Calculate the value of the following sum for n taking values 4, 8, 12 and n. What is
the limiting value of the sum as n gets large?

> seq(Sum((1/2)^i,i=0..4*j)=sum((1/2)^i,i=0..4*j),j=1..3);

4
∑

i=0

(
1

2
)i =

31

16
,

8
∑

i=0

(
1

2
)i =

511

256
,

12
∑

i=0

(
1

2
)i =

8191

4096

> Sum((1/2)^i,i=0..n)=sum((1/2)^i,i=0..n);

n
∑

i=0

(
1

2
)i = −2 (

1

2
)(n+1) + 2

As n gets large, ( 1
2 )n goes to 0, so the whole sum goes to 2.

> Sum((1/2)^i,i=0..infinity)
=limit(Sum((1/2)^i,i=0..n),n=infinity);

∞
∑

i=0

(
1

2
)i = lim

n→∞

n
∑

i=0

(
1

2
)i

7.2 Riemann Sums with the student package

There are some useful words in the student package dealing with integration.
> with(student);

[D, Diff , Doubleint , Int , Limit , Lineint , Product , Sum, Tripleint , changevar ,

completesquare , distance , equate, integrand , intercept , intparts , leftbox , leftsum,

makeproc, middlebox , middlesum , midpoint , powsubs , rightbox , rightsum ,

showtangent , simpson , slope , summand , trapezoid ]
For example, middlesum and middlebox compute and display a regular Riemann sum where

the mark is chosen as the midpoint in each subinterval. For example, to get the middle regular
Riemann sum for xˆ2 from 0 to 1 with 10 subintervals,

> middlesum(x^2,x=0..1,10) = value(middlesum(x^2,x=0..1,10));

1

10

(

9
∑

i=0

(
i

10
+

1

20
)2
)

=
133

400

> middlebox(x^2,x=0..1,10);

0

0.2

0.4

0.6

0.8

1

0.2 0.4 0.6 0.8 1
x

Two Area Problems
Exercise: Plot the ellipse x2

4 + y2 = 1 with a parametric plot. Estimate the area of the ellipse.

Use the fact that it is four times the area under the graph of
√

1 − x2

4 for x between 0 and 2.
Solution:

A parameterization of the ellipse is to let x = 2 cos(t), y = sin(t), as t goes from 0 to 2π.
> ellipse := [2*cos(t),sin(t),t=0..2*Pi];

ellipse := [2 cos(t), sin(t), t = 0..2π]

> plot(ellipse,scaling=CONSTRAINED);

–1

–0.5

0.5

1

–2 –1 1 2

The upper ellipse is the graph of the function
> f := x -> sqrt(1- x^2/4);

f := x →
√

1 − 1

4
x2

We can use leftsum and rightsum to estimate the area.
> for i from 1 to 5 do

left := 4*evalf(value(leftsum(f(x),x=0..2,25*i)));
right := 4*evalf(value(rightsum(f(x),x=0..2,25*i)));
print(left,right,right-left); od:

6.424392628, 6.104392628, −0.320000000

6.356537020, 6.196537020, −0.160000000

6.332899024, 6.226232356, −0.106666668

6.320834060, 6.240834060, −0.080000000

6.313502780, 6.249502780, −0.064000000

By inspecting the table of numbers, we see that area is between 6.24 and 6.31.

Exercise: For the function 10 − x4 + 10x2, plot the function and determine the region R where
the graph of f lies above the x-axis. Make the sketch of R and compute the endpoints of the base
of R. Use student[middlesum] to find the area of the region R to within 0.1.

Solution:
First plot the function and calculate the endpoints of R.
> y := 10 - x^4 + 10*x;

y := 10 − x4 + 10x

> plot(y,x=-2..3);

–40

–30

–20

–10

0

20

–2 –1 1 2 3x

> sol := fsolve(y,x);

sol := −0.9263593057, 2.417910164
> for i from 1 to 5 do

value(student[middlesum](y,x=sol[1]..sol[2],20*i))
od;

41.78810165

41.73598953

41.72632972

41.72294809

41.72138277

The area looks to be 41.7 to the nearest 0.1, using middlesum with 100 subdivisions.

7.3 Learning to use Int and int.

The fundamental theorem of calculus tells us that a definite integral of a continuous
function f can be evaluated by first finding an antiderivative of f and then calculating the

difference of its values at the endpoints: In symbols,

∫ b
a f(x) dx = F(b) − F(a)

provided d
dx F(x) = f(x).

You can use the Maple word int to find antiderivatives, (aka indefinite integrals).
For example, to find an antiderivative of x2 + cos(x

5 ), use int.
> Int(x^2 + cos(x/5),x)=int(x^2 + cos(x/5),x);

∫

x2 + cos(
x

5
) dx =

x3

3
+ 5 sin(

x

5
)

Notice that there is a passive form of int, Int, for typesetting purposes.
You can also use int to evaluate definite integrals. For example, if we want to plot the region

under the graph of x2 + cos(x
5 ) for x between 0 and 2 and calculate its area.

> restart;

> f := x-> x^2 + cos(x/5);

f := x → x2 + cos(
1

5
x)

> Int(f(x),x=0..2)=int(f(x),x=0..2);

∫ 2

0
x2 + cos(

x

5
) dx =

8

3
+ 5 sin(

2

5
)

> area := evalf(int(f(x),x=0..2));

area := 4.613758379

We will use plots[polygonplot] to shade the region.
> ### WARNING: semantics of type ‘string‘ have changed

plots[polygonplot]( [ [2,0],[0,0],[0,f(0)],seq([x/10,f(x/10)] ,
x=0..20) ],color=gray,title= cat(‘Area = ‘,convert(area,string)) );

Area = 4.613758379

5

0.5 1 1.5 2

Exercises:
Exercise: For each of the functions below, Sketch the region under the graph and calculate the

area of the region using the fundamental theorem of calculus.
a) f(x) = (1 − x + x2)2 , for x = −1..2

b) Under the arch of f(x) = cos( x
2 )2 + 3x which contains x=0. .

c)
> f(x) = piecewise( x<1, sqrt(x) ,x>=1,x^2);

for x between 0 and 2.

x2 + cos(
x

5
) =

{√
x x < 1

x2 1 ≤ x

7.4 Average value.

Suppose you picked at random 1000 numbers between 0 and 1. What would you expect the average
of the squares of these numbers to be, about?

Solution.
First let’s look up how rand works.

> ?rand

So to get a function which returns random numbers between 0 and 1, we could do something
like this:

> f := rand(10^7)/10.^7:

> f();

0.9669081000
> s := 0:

for i from 1 to 1000 do
s := s+f()^2 od:
s/1000;

0.3257097986

Looks like its around a third.
On the other hand, you could estimate the average value by averaging the values at n equally

spaced points in the interval. Here is a procedure for doing this. The inputs are f, the function to
be averaged; a,b, the endpoints of the interval where the function is evaluated; and n, the number
of equally spaced values to be averaged.

> av := proc(f,a,b,n)
local i,dx;
dx := (b-a)/n;

> evalf(sum(f(a+i*dx),i=1..n)/n);
end:

So to estimate the average value of xˆ2 as x ranges from 0 to 1, we could generate a sequence
of estimates

> for i by 100 from 100 to 400 do
print(i, av(x->x^2,0,1,i)) od;

100, 0.3383500000

200, 0.3358375000

300, 0.3350018519

400, 0.3345843750

This seems to give about 1/3 also.
The last estimate can be turned into a regular Riemann sum. Let ∆x = b−a

n denote the length
of each subinterval in the regular partition of [a,b] into n subintervals. Then the average and
Riemann sum are the same:
∑n

i=1
f(a+i ∆x)

n =

∑

f(a+i ∆ x) ∆x

b−a

As n gets large, this Riemann sum converges to the integral of the function over the interval
divided by the length of the interval. Thus we are justified in defining the average value as an
integral.

Problems.
Exercise: The temperature in the 24 period from 0..24 is given as f(t) = 10 sin( t

3 ) + 100 t (24 − t).
Find the average temperature over the time interval. Find a time t when it is the average temper-
ature. Also find the maximum temperature and minimum temperature.

Exercise: Suppose f is a continuous function on [a,b], with average value T. Show that there is
an x in [a,b] so that f(x) = T.

7.5 Modeling the flow of air in lungs:

The rate of change of the volume of air in the lungs can be modeled very roughly (according to
some texts) by the function

> f := t -> 1/2*sin(2*Pi*t/5);

f := t → 1

2
sin(

2

5
π t)

where volume is measured in liters and time is measured in seconds. So the actual volume of
air in the lungs would obtained by integrating f. Here we are assuming there is no air in the lungs
at time 0.

> Int(f(tau),tau=0..t) = int(f(tau),tau=0..t);

∫ t

0

1

2
sin(

2π τ

5
) dτ = −5

4

−1 + cos(
2π t

5
)

π
> F := unapply(int(f(tau),tau=0..t),t);

F := t → −5

4

−1 + cos(
2

5
π t)

π
Let’s plot the rate of airflow and the volume function to get a feel for their behavior.
> plot({f,F},0..10);

Questions.
Question: Which is the rate of airflow function and which is its integral?

Question: What is the maximum volume of air in the lungs in this model? The minimum
volume?

Question: When are the lungs half-full?
Question: What is the average rate of airflow in the lungs over the time interval [0,2.5]? [2.5,5]?

[0,5]?
> ar := evalf(int(f(t),t=0..2.5)/2.5);

ar := 0.3183098862

About .32 liters per second on average.
Question: What is the average volume of air in the lungs over the time interval [0,2.5]? [0,5]?
> av := evalf(int(F(t),t=0..2.5)/2.5);

av := 0.3978873577

About .4 liter air in the lungs on average.
Actually, there are different, perhaps more realistic models of the rate of airflow into the lungs.

For example, consider this one,
> g := t -> (3/8-1/4*cos(4*Pi*t/5))*sin(2*Pi*t/5);

g := t → (
3

8
− 1

4
cos(

4

5
π t)) sin(

2

5
π t)

Problem: Using the function g to model the rate of airflow in the lungs, find the volume G of
air in the lungs at time t. (Assume as before that G(0) = 0.) Plot both g and G over the time
interval [0,10]. Describe qualitatively a difference that you note between this model and the model
we looked at above. Calculate the maximum rate of airflow in this model. What is the maximum
volume of air in the lungs in this model? What is the average volume of air in the tank in the time
interval [0,5]

7.6 Two Area problems.

Exercise: Find the upside down parabola y = a − x2 so that the area between it and the parabola
y = x2 is 100. Draw a diagram.

Solution.
Set up the functions which bound the region.
> restart;

> f := x -> a-x^2;

f := x → a − x2

> g := x-> x^2;

g := x → x2

Find the points where the functions cross.
> sol := solve(f(x)=g(x),x);

sol :=

√
2
√

a

2
, −

√
2
√

a

2

Get the area of the region. By inspection, we see that sol[2] is the left endpoint where the
functions cross.

> area := int(a-2*x^2,x=sol[2]..sol[1]);

area :=
2 a(3/2)

√
2

3
> sol;

√
2
√

a

2
, −

√
2
√

a

2
Solve an equation.
> sol2 := fsolve(area=100,{a});

sol2 := {a = 22.40702373}
> assign(sol2);

> plot({f ,g }, sol[2]..sol[1]);

Another area problem: The parabola y = a x2 + b x + c is tangent to the graph of y = 2 + |x − 3|
at two points and the area of the region bounded by their graphs is 10. Find a, b, and c. Make a
sketch.

Solution:
The axis of the parabola is x = − b

2 a . That is also the axis of y = 2 + |x − 3| , so − b
2 a = 3, or

b = −6 a . The point where the slope of the parabola is 1 is on both graphs. Call the point [x0,y0].
Then x0 − 1 = a x0 2 − 6 a x0 + c and 1 = 2 a x0 − 6 a.

> restart;

> eq1 := x0-1 = a*x0^2 -6*a*x0 + c;

eq1 := x0 − 1 = a x0 2 − 6 a x0 + c

> eq2 := 1 = 2*a*x0 - 6*a;

eq2 := 1 = 2 a x0 − 6 a

> ac := solve({eq1,eq2},{a,c});

ac := {c =
x0 2 − 2 x0 + 6

2 (x0 − 3)
, a =

1

2 (x0 − 3)
}

Finally, the area between the curves is 100, so the righthand half is 50.
> eq3 := Int (a*x^2-6*a*x+c-(2+x-3),x=3..x0)=50;

eq3 :=

∫ x0

3
a x2 − 6 a x + c + 1 − x dx = 50

Integrating,.
> eq3 :=int(a*x^2-6*a*x+c-(2+x-3),x=3..x0)=50;

Sollving for x0
> sol :=solve(subs(ac,eq3),x0);

sol := 3 + 10
√

3, 3 − 10
√

3
> assign({x0=sol[1]});

assign(ac);

> plot({2+abs(x-3),a*x^2-6*a*x+c},
x=sol[2]-2..sol[1]+2,color=black);

8 Moments and Center of Mass

8.1 Center of mass of a Wire

Suppose we have a wire l feet long whose density is ρ(x) pounds per foot at the point x feet from
the left hand end of the wire. What is the total mass of the wire and where is its center of mass
, i.e., the point cm about which the total moment of the wire is 0?

• Mass Chop the wire into n small pieces each ∆xi feet long and pick an arbitrary point ci

in each piece. An approximation to the mass of the ith piece of wire is ρ(ci)∆xi, so an
approximation to the total mass is

∑n
i=1 ρ(ci)∆xi . This approximate mass is a Riemann

sum approximating the integral
∫ l
0 ρ(x) dx, and so the mass of the wire is defined as the value

of this integral.

• Center of mass: Chopping as above, the approximate moment of the ith piece about the
center of mass cm is (ci − cm) ρ(ci)∆xi and so the total approximate moment is

∑n
i=1 (ci − cm) ρ(ci)∆xi.

This is seen to be a Riemann sum approximating the integral
∫ l
0(x − cm) ρ(x) dx. But the

center of mass is defined as the point about which the total moment is zero so the integral
satisfies the equation

∫ l
0(x − cm) ρ(x) dx = 0. Using properties of integrals, we can solve this

equation for cm, to get the ratio of integrals cm =

∫ l

0
xρ(x) dx

∫ l

0
ρ(x) dx

. Note the top integral represents

the total moment of the wire about its left end (x=0) and the bottom integral is the total
mass of the wire.

Exercise: Find the center of mass of a wire 1 foot long whose density at a point x inches from
the left end is 10 + x + sin(x) lbs/inch.

8.2 Center of mass of a solid of revolution

If 0 ≤ y = f(x) for a ≤ x<= b, then let S be the solid of revolution obtained by rotating the
region under the graph of f around the x axis. We know how to express the volume of S as an integral:
Just integrate from a to b the crossectional area π f(x)2 of the solid S to get Volume =

∫ b
a π f(x)2 dx

Now how would we find the center of mass of the solid, assuming it’s made of a homogeneous
material? Well, it’s clear that the center of mass will be somewhere along the x-axis between a and
b. Let CM be the center of mass. Partition [a, b] into n subintervals [xi, xi+1] and using planes
perpendicular to [a, b] approximate the solid S with the n disks where the ith one has volume
∆Vi = π f(xi)

2 ∆xi

Now the signed moment of the ith disk about the point CM is Mi = (x − CM )∆Vi and the
sum of these moments will be approximately 0, since CM is the center of mass. If we let ∆xi go
to zero this approximate equation becomes an equation for the center of mass:

> CMequation := int((x-CM)*Pi*f(x)^2,x=a..b) =0;

CMequation :=

∫ b

a
(x − CM )π f(x)2 dx = 0

Useing properties of integrals, we can solve this equation for CM.
> sol := solve(CMequation,{CM} ) ;

sol :=



















CM =

∫ b

a
f(x)2 x dx

∫ b

a
f(x)2 dx



















Notice that the center of mass of the solid of revolution is the same as the center of mass of a
wire whose density at x is the area of the cross-section.

We can define a word cenmass which takes a function f, an interval [a,b], and locates the center
of the solid of revolution.

> cenmass := proc(f,a,b)
int(x*f(x)^2,x=a..b)/int(f(x)^2,x=a..b) end:

For example, the center of the solid obtained by rotating the region R under the graph of
y = cos(x) for x between 0 and π

2 is
> cenmass(cos,0,Pi/2);

4

1

16
π2 − 1

4
π

Now we can define a word to draw the solid and locate the center of mass.
> drawit := proc(f,a,b)

local cm, solid;
### WARNING: the definition of the type ‘symbol‘ has changed’; see

> help page for details
cm :=
plots[pointplot3d]([evalf(cenmass(f,a,b)),0,0],color=red,symbol=box,th

> ickness=3):
solid :=
plots[tubeplot]([x,0,0],x=a..b,radius=f(x),numpoints=20,style=wirefram

> e);
plots[display]([cm,solid],scaling=constrained); end:

Test this out.

> drawit(cos+2,0,7);

We can animate the motion of the center of mass as the solid changes.

> plots[display]([seq(drawit(2+cos,0,t),t=1..10)],
insequence=true);

Exercise: Find the center of mass of a homogeneous hemispherical solid.
Exercise: A homogeneous solid is in the shape of a parabolic solid of revolution obtained by

rotating the graph of y = xˆ2 , x in [0,a] around the the y axis, for some positive number a. If the
center of mass is at y= 10, what’s a?

9 Definitions and Theorems of Calculus I

The theorems of Calculus up to the fundamental theorem fit into the nice logical sequence given
below. The following diagram gives a partial picture of how the theorems are related.

FDT

SDT

Antdef

 

SST

TOA FUND

FST

EXV1

EXV2

IV

Conprps

RULD

DIC

MV

IMV
Intdef

Limdef Condef

LUB

Derdef

Limprps

CII

Intprps

[Limdef] Definition of Limit. Let a be a number and let f be a function defined at least
on an open interval which contains a or has a as an endpoint. The limit of f at a is L means that
for any positive number ε there is a positive number δ such that for each number x at which f is
defined and which is within δ of a (but different from a ), f(x) is within ε of L .

[Limprps]
Sum and Product of Limits. If each of two functions have a limit at a point, then so

does their sum, difference, and product; in fact, the limit of the sum (difference, product) is the
sum (difference, product) of the limits. Also the limit of the quotient is the quotient of the limits,
provided the limit of the denominator is not 0.

Hint on proof: Use the definition of limit to prove these properties.
The Squeeze Theorem. If each of two functions has the same limit at a point, and a third

function is squeezed between them, then it also has that limit at the point.
Hint on proof: Use the definition of limit.
[Condef] Definition of continuous function. A function is continuous at a point if it is

defined there and its value there is its limit there. A function is continuous if it is continuous at
each point where it is defined.

[Contprps] Sum, Product, of Continuous Functions. The sum (difference, product) of
two continuous functions is continuous, at each point where they are both defined. The quotient of
two continuous functions is continuous at each point where they are both defined and the denom-
inator is not 0. The composition of two continuous functions is continuous at each point where it
is defined.

Hint on proof: Use SPLT.
[Derdef] Definition of Derivative. A function f is differentiable at a provided it is defined

at a and the limit of f(a)−f(x)
a−x at a exists. The derivative f is the function D(f) whose value at a is

the above limit, if it exists.
[RULD] Rules for Derivatives. The derivative of a constant function is the constant function

0. The derivative of the sum (difference) of two functions is the sum (difference) of their derivatives.
The derivative of the product of two functions is the derivative of the first times the second plus

the first times the derivative of the second. The derivative of the quotient of two functions is the
quotient of the derivative of the top times the derivative of the bottom minus the top times the
derivative of the bottom with the square of the bottom. The derivative of the composition of two
functions is the derivative of the first composed with the second times the derivative of the second.

Hint on proof: Use the definition of derivative.
[DIC ] Differentiability Implies Continuity. If a function is differentiable at a point, then

it is continuous at that point.
Hint on Proof: Use the definitions.
[IV] Intermediate Value Theorem. If a function is continuous on an interval,then each

number between two values of the function is itself a value of the function.
Hint on proof: You need the least upper bound property to prove this theorem.
[EXV1] Extreme Value I. If a function is continuous on a closed interval, then it has both

a minimum value and a maximum value somewhere in the interval.
Hint on Proof: You need the least upper bound property to prove this theorem.
[EXV2] Extreme Value II . If a function is continuous on a closed interval, then its extreme

values on the interval occur at the endpoints of the interval or at the places interior to the interval
where the derivative is 0 or not defined.

Hint on Proof: Use the definition of continuous function.
[ROLL] Rolle’s Theorem . If a function is continuous on a closed interval, differentiable at

each point inside the interval, and has the same value at the endpoints of the interval, then its
derivative is 0 at some point inside the interval.

Hint on Proof: Use the extreme value theorems.
[MVT] Mean Value Theorem. If a function is continuous on a closed interval and differ-

entiable at each point inside the interval, then at some point inside the interval the derivative is
equal to the average change in the function over the interval.

Hint on Proof: Use Rolle’s Theorem.
[FST] First Sign Theorem. If the derivative of a function is positive (negative) at each point

in an interval, then the function is increasing (decreasing) over the interval.
Hint on Proof: Use the Mean Value Theorem.
[FDT] First Derivative Test. If the sign of the first derivative changes from positive to

negative (negative to positive) across a critical point, then the function has a local maximum value
(local minimum value) at the critical point.

Hint on Proof: Use the first sign theorem.
[SST] Second Sign Theorem. If the second derivative of a function is positive (negative) on

an interval , then the function is concave up (down) on that interval.
Hint on Proof: Use the Mean Value Theorem.
[SDT] Second Derivative Test. If the second derivative of a function is positive (negative)

at a critical point of the function, then the function has a local minimum (maximum) at the point.
No conclusion may be drawn if the second derivative is 0 at the critical point.

Hint on Proof: Use the second sign theorem.
Definition of Antiderivative. A function F is called an antiderivative of a function f provided

the derivative of F is f .
[TOA] Theorem on Antiderivatives. If two functions have the same derivative on an

interval then they differ by a constant function on that interval.
Hint on Proof: The proof of this theorem uses the Mean Value Theorem.

Definition of Integral. Let f be defined on [a,b] . A partition P of [a,b] is a finite set of points
in [a,b] , a = x0, xi < xi+1, xn = b for i from 0 to n. [xi, xi+1] is a subinterval of the partition.
The mesh of the partition is the width of the longest subinterval of the partition. A Riemann Sum
of f over the partition P is a number

∑n−1
i=0 f(xi

∗) (xi+1 − xi) where each xi
∗is chosen from the

subinterval [xi, xi+1] .The integral of f over [a,b] is the number L such that for any positive number
ε , there is a positive number δ such that each Riemann Sum of f over each partition of mesh less
than δ is within ε of L , provided such a numberL exists.

[FET] First Evaluation Theorem. The integral of a constant function over an interval is
the constant times the width of the interval.

Hint on Proof: Use the definition of the integral.
[LPR] Linearity Property. The sum of two functions, each integrable over an interval, is

itself integrable over the interval, and the integral of the sum is the sum of the integrals. The
integral of a constant times a function is the constant times the integral of the function.

Hint on Proof: Use the definition of integral.
[APR] Additivity Property. If a function is integrable over an interval, and the interval is

partitioned into two subintervals, then the integral of the function over the entire interval is the
sum of the integrals of the function over the subintervals.

Hint on Proof: Use the definition of integral.
[CPR] Comparison Property. If the graph of one function never goes above the graph of a

second function, then the integral of the first function does not exceed the integral of the second
function.

Hint on Proof: Use the definition of integral.
[CIT] Continuity Implies Integrability. If a function is continuous on an interval, then it

is integrable over the interval.
Hint on Proof: The proof of this theorem requires the least upper bound property of the real

numbers.
[PPR] Positivity Property. If a function is continuous and non-negative on an interval, and

takes on a positive value, then its integral over the interval is positive.
Hint on Proof: Use the comparison property and the definition of continuous function.
[IMV] Integral Mean Value Theorem. If a function is continuous on an interval, then the

value of its integral over the interval is the value of the function someplace in the interval times
the width of the interval.

Hint on Proof: The proof of the integral mean value theorem uses the first extreme value
theorem, the first evaluation theorem, the comparison property, and the intermediate value theorem.

[FUND] Fundamental Theorem. If a function is continuous on an interval, then (i) (exis-
tence) it has an antiderivative on the interval; in fact, the function F defined on [a,b] by the formula
F(x) = the integral of f over the interval [a,x] is an antiderivative of f on [a,b] .Furthermore, (ii)
(Evaluation) the integral of the function over the interval is obtained by taking any antiderivative
of the function, and subtracting its value at the left-hand endpoint from its value at the right-hand
endpoint.

Hint on Proof: To prove (i), use the integral mean value theorem. To prove (ii), use (i) and the
theorem on antiderivatives above.

10 Inverse Functions

10.1 A Useful Function – The natural logarithm

The logarithm function was first introduced in the early 1600’s by John Napier to reduce the toil
of performing multiplication and division of numbers. The idea was quickly accepted worldwide.

Napier’s logarithm was defined differently from the way it would be now. The idea of defining
the logarithm of x for x > 0 to be the signed area under the curve 1/x, that is, ln(x) =

∫ x
1

1
t dt ,

came along later. The key fact is that this area grows arithmetically as x grows geometrically.
Problem: Consider the sequence of powers of 2, an = 2n , n = 1, 2, .... This is a geometric

sequence , that is, the ratio of consecutive terms is constant (in this case, the common ratio is 2).
Now let bn =

∫ an
1

1
t dt, for n = 1, 2 .. We claim that the sequence bn is an arithmetic sequence

, that is, the difference of consecutive terms is constant. To compute the difference, first use the

definition: bn+1 − bn =
∫ 2(n+1)

1
1
t dt −

∫ 2n

1
1
t dt. This difference of integrals can be written as the

single integral
∫ 2(n+1)

2n
1
t dt. But now make the substitution t = u

2n into the integral. Then dt = 1 du
2n

and the integral becomes
∫ 2
1

1
u du = ln(2). It follows from this that bn = n ln(2), an arithmetic

sequence as we claimed above.

Exercise: Use the change of variable formula, given below, to establish other properties of the
natural logarithm, such as i) ln(x y) = ln(x) + ln(y) , for all positive x and y, and ii) ln( 1

x) = −ln(x)
, for all x > 0.

∫ b
a f(g(x))D(g)(x) dx =

∫ g(b)
g(a) f(u) du

Exercise: Use a sequence of student[rightsum]’s to estimate ln(2) by Riemann sums to 4 decimal
places.

Exercise: Which line y = m (x − 1) divides the region under the graph of y = 1
x between 1 and

2 into two regions with equal area? Sketch the graphs.
Solution:
First find where the line crosses 1/x.
> y := m*(x-1);

y := m (x − 1)

> b := solve(y=1/x,x);

b :=
1

2

m +
√

m2 + 4m

m
,

1

2

m −
√

m2 + 4m

m
We want the first solution. That’s the one that crosses the right branch of 1/x. The equation

we want to solve then is:
> eq := Int(1/x-y,x=1..b[1])=1/2*ln(2);

83



eq :=

∫ 1/2
m +

√
m2 + 4m

m
1

1

x
− m (x − 1) dx =

1

2
ln(2)

> m := fsolve(value(eq),m);

m := 0.7575012634

Now go ahead and draw the picture to see if it looks about right.
> plot({1/x,y,[[2,0],[2,1/2]]},x=1..2,scaling =

constrained,color=black);

This looks to be cut in half.
Exercise: Find the parabola y = x2 + b which is tangent to the graph of y = ln(x) . Sketch the

graph.
Solution:
Let [x1,ln(x1)] be the point of tangency. Then ln(x1) is equal to x1ˆ2 + b. That’s the first

equation.
> restart;

eq1 := x1^2+b = ln(x1);

eq1 := x1 2 + b = ln(x1 )

But also the derivative of xˆ2 + b at x1 and 1/x at x1 must be equal. Thats the second
equation, which should be enough to determine x1 and b.

> eq2 := 2*x1 = 1/x1;

eq2 := 2 x1 =
1

x1
> sol := solve({eq1,eq2},{x1,b});

sol := {b = −1

2
+ ln(RootOf(2 Z 2 − 1)), x1 = RootOf(2 Z 2 − 1)}

There seem to be two values for x1, 1/sqrt(2) and -1/sqrt(2). But the ln is not defined at
-1/sqrt(2) so x1 is the first one.

> x1 := 1/sqrt(2);

x1 :=
1

2

√
2

> b := ln(x1)-1/2;

b := ln(
1

2

√
2) − 1

2

> plot({ln(x),x^2+b},x=-2..2,thickness=2);

x

Exercise: Find the parabola y = a x2 which is tangent to the graph of y = 1
x . Sketch the

graphs.

Exercise: Find the local extreme values of y = ln(10 + x (x − 1) (x − 2)) . Sketch the graph.

Exercise: Sketch the region bounded by y = ln(1 + x2) and y = 10 − x2. Find the area of the
region.

8. The line x + y = m and the graph of y = 1
x bound a region of area 10. Find m and plot the

graphs.

Exercise: The line y = 2x + 1 does not intersect the graph of y = ln(ln(x)) . (Does it?) Find
the point on the line which is closest to the graph. Sketch the graphs.

Solution:
First look at the graphs.
> line := 2*x+1;

line := 2x + 1

> lnln := ln(ln(x));

lnln := ln(ln(x))

> plot( {line,lnln} ,x=-5..5,thickness=2);

Clearly they don’t intersect. The lnln function is only defined for x > 1 and is concave down,
so it will never cross the line. The closest point looks to be near x=1.5 It would be the point on the
graph where the tangent line is parallel with the given line. So it would be where the derivative is
2.

> x1:=fsolve(diff(lnln,x)=2,x,1..1.5);

x1 := 1.421529936
> plot({line,lnln,-1/2*(x-x1)+subs(x=x1,lnln)

},x=-5..5,scaling=constrained,thickness=2,color=black);

10.2 The inverse of the natural log

The inverse of the natural logarithm is called the exponential function.
It is predefined in Maple as exp. It is called the exponential function because it obeys the ’laws

of exponents’,
e(x+y) = ex ey

and e(−x) = 1
ex . In this setting, e is defined to be the number such that ln(e) = 1 . This says

that exp(1)=e. So, exp(2) = exp(1+1) = exp(1)*exp(1) = eˆ2 . Also, exp(1/2) = eˆ(1/2) , since
exp(1/2)ˆ2 = exp(1/2)*exp(1/2) = exp(1/2 + 1/2) = e . With some work, you can show that
exp(x) = eˆx for any rational number x, and so it is compatible to define eˆx = exp(x) for all real
numbers x.

The number e can also be shown to be the limit of (1 + 1
n)n as n gets large. We can check this

out numerically by defining a function which takes as input a positive integer n and outputs the
decimal value of the nth term of the sequence.

Exercise: By experimenting, find the smallest value of n which gives e correct to 5 significant
digits.

> approxe := n -> evalf((1+1/n)^n);

approxe := n → evalf((1 +
1

n
)n)

> approxe(100);

2.704813829

The exponential function rises more quickly than any polynomial function.
The linear function which has the same value and same derivative as the exponential function

y = exp(x) at x = 0 is the tangent line y = x + 1 at x=0. The quadratic function that has the
same value, and first and second derivatives as exp(x) at x = 0 is the quadratic function 1 + x +
(1/2)xˆ2 . Continuing, the polynomial of degree n which has the same value and first n derivatives

as the exponential function can be shown to be
∑n

i=0
xi

n! . We can plot any number of these on the
same graph with exp(x).

> efun := plot(exp,-1..4,color=black,thickness=2):

> pfuns := plot({1+x,1+x+1/2*x^2,1+x+x^2/2+x^3/3!,
1+x+x^2/2+x^3/3!+x^4/4!,1+x+x^2/2+x^3/3!+x^4/4!+x^5/5!
},x=-1..4,color=tan):

> plots[display]([efun,pfuns]);

As can be seen from the plots, the polynomials get closer to the exponential function but not rise
as fast. These polynomials are the Taylor approximating polynomials for the exponential function,
and can be formed for any function which has derivatives of all orders.

The functions y = xˆ2 and y = 2ˆx cross each other 3 times. Find
two of the points of intersection by inspection of the equation xˆ2 = 2ˆx . Then plot the two

functions on the same graph and estimate, from the graph, the coordinates of the third point.
The exponential and logarithm functions can be used to define the other exponential functions

ax = e(x ln(a)), so long as a is positive.

Exercise: The functions 3x and x3 touch each other 2 times. Find one of the point of intersection
by inspection of the equation 3x = x3. Then plot the two functions on the same graph and estimate,
from the graph, the coordinates of the second point. Find the value of a between 2 and 3 such that
ax and xa touch exactly once. Plot the resulting graphs.

Two cups o’ soup: Suppose that two cups of soup, the first at 90 C and the second at 100 C
are put in a room where the temp is maintained at 20 C. The first cup cooled from 90 C to 60 C
after 10 minutes, at which time the second cup is put in a freezer at -5 C. How much longer will it
take the two cups to reach the same temperature?

Solution:
Assuming Newton’s Law of cooling, we see that the first cup’s temperature is given by
> restart;

> f := t -> 20 + 70*exp(k*t);

f := t → 20 + 70 e(k t)

where k is determined by the information given that the first cup cools to 60 C in 10 minutes
(i.e. solve the equation f(10)=60 for k).

> k := solve(f(10)=60,k);

k :=
1

10
ln(

4

7
)

Assume that this value of k is also valid for the second cup. (Why shouldn’t it be, they are
both soup.) So the second cup’s temperature is given by

> g := t -> 20 + 80*exp(k*t);

g := t → 20 + 80 e(k t)

for the first 10 minutes. From that time on, the second cup’s temperature is given by
> h := t -> -5 + (itemp - (-5))*exp(k*t);

h := t → −5 + (itemp + 5) e(k t)

where itemp is the temperature that the 2nd cup would start at in order for it to cool to g(10)
degrees C if it were in the freezer from the start. So itemp satisfies the equation

We can combine these two rules into one function using piecewise.
> c2 := unapply(piecewise(t<=10,g(t),t>=10, h(t)),t);

c2 := t → piecewise(t ≤ 10, 20 + 80 e(1/10 ln(4/7) t), 10 ≤ t, −5 +
495

4
e(1/10 ln(4/7) t))

> eq := -5 + (itemp +5)*exp(k*10) = g(10);

eq := −15

7
+

4

7
itemp =

460

7

Solving for itemp,
> itemp := solve(eq,itemp);

itemp :=
475

4

Now plot the two temperature curves.
> cup1 := [t,f(t),t=0..20];

cup1 := [t, 20 + 70 e(1/10 ln(4/7) t), t = 0..20]

> cup2 := [t,c2(t),t=0..20];

cup2 :=



t,







20 + 80 e(1/10 ln(4/7) t) t ≤ 10

−5 +
495

4
e(1/10 ln(4/7) t) 10 ≤ t

, t = 0..20





> plot( {cup1,cup2},thickness=2 );

The cups are at the same temperature before another 5 minutes pass.

> etime := solve(f(t)=c2(t),t);

etime := 10
ln(

20

43
)

ln(
4

7
)

> evalf(etime),‘ minutes from time 0‘;

13.67845330, minutes from time 0

At that time the common temperature is
> f(etime)=evalf(f(etime)),‘ degrees‘;

2260

43
= 52.55813953, degrees

10.3 Inverse Functions: The inverse trig functions

None of the trignometric functions are 1 to 1, however each can be restricted to a domain so that
it is 1-1 on that domain and all values of the function are attained. For example, if we restrict
the tangent function to the interval (-Pi/2 , Pi/2), it is 1-1 there. So the inverse tangent or arctan
function is defined as the inverse of the restriction. The same sort of thing is done to define the
arcsin and arcsec functions.

> Digits:=3:
plots[display](matrix(1,3,[[plot(arctan ,-10..10,thickness=2),
plot(arcsin ,-10..10 ,thickness=2),

> plot(arcsec ,-10..10,thickness=2)]]));
Digits := 10:

2.40

2.20
2.00

1.80

1.60

1.40
1.20

1.00

.800

10.5.0–5.–10.

.80

.60

.40

.20
0

-.200
-.400
-.600
-.800

.80.60.40.200-.200-.400-.600-.800

1.00

.50

0

-.50

–1.

10.5.0–5.–10.

One thing which makes the inverse trig functions especially important is that they all have
algebraic derivatives. So they play an important role in finding antiderivatives of functions.

> restart;

> diff(arctan(x),x);

1

1 + x2

> plot({arctan,D(arctan)},-5..5,thickness=2);

Exercise: Find the critical points of arctan( 18
x ) − arctan( 3

x) , for positive x. Make a sketch
which shows any extreme values.

> y := arctan(18/x)-arctan(3/x);

y := arctan(
18

x
) − arctan(

3

x
)

> plot(y,x=-20..20);

By inspection, we can be pretty sure that y has a single max value at around [7,.8] and a
symmetrically placed min at around [-7,-.8]. To be more specifiy

> extr :=solve(diff(y,x),{x});
extr := {x = 3

√
6}, {x = −3

√
6}

There is a maximum at 3
√

6 and a minimum at −3
√

6 .
Exercise: Bill, driving a red Mustang at 120 miles per hour, is headed straight toward a railroad

crossing in Kansas. When he is a mile from the crossing he notices a train, coming towards the
same intersection. He estimates the speed of the train to be 60 mph and the distance of the engine
from the intersection to be a mile. The train looks to be about 300 yards long. Keeping his eye on
the train during the next minute, he notices that the angle subtended at his eye by the train got
larger at first and then began decreasing. How far was Bill from the intersection when the angle
was maximum?

Solution:
First, draw a picture. Let’s use yards, minutes, and degrees for units. Then Bill’s distance from

the intersection t minutes from the time Bill first saw the train was y = 1762 − 2 (1762) t yards and
the engine’s distance was x = y yards. The distance of the caboose from the intersection is x + 300
yards. The angle subtended at Bills eye by the train is β − α Now tan(β) = x+300

y and tan(α) = x
y

.
So the function to be maximized is ang = arctan( x+300

y ) − arctan(x
y ) .

> restart;

> ang := arctan((x+300)/y)-arctan(x/y);

ang := arctan(
x + 300

y
) − arctan(

x

y
)

> y := 1762*(1-2*t); x:= 1762-1762*t; ang;

y := 1762 − 3524 t

x := 1762 − 1762 t

arctan(
2062 − 1762 t

1762 − 3524 t
) − arctan(

1762 − 1762 t

1762 − 3524 t
)

> plot(ang,t=-2..3);

The plot shows that there is a local maximum viewing angle before Bill gets to the crossing, and
an absolute maximum after he crosses. To get the times, distances, and angles do the following:

> sol :=solve(diff(ang,t),t) ;

sol :=
1

2
+

1

8810

√
5202305,

1

2
− 1

8810

√
5202305

> evalf([sol]);

[0.7588940067, 0.2411059933]

> asol :=evalf([seq(subs(t=sol[i],ang),i=1..2)]);

asol := [−0.2355710897, 0.0903370978]

> dsol :=evalf([seq(subs(t=sol[i],[x,y]),i=1..2)]);

dsol := [[424.8287602, −912.3424796], [1337.171240, 912.3424796]]

> evalf([seq(convert(asol[i],degrees),i=1..2)]);

[−13.49722921 degrees , 5.175934435 degrees ]

So Bill’s view of the train is locally a maximum of about 5 degrees at about .24 minutes after
he sees the train and about 912 yards before he crosses the intersection. His maximum view (about
13.5 degrees) of the train occurs at about .76 minutes and 912 yards beyond the crossing.

Exercise: Find an antiderivative for
∫

a

b + c x2
dx

> int(a/(b+c*x^2),x);

arctan(
c x√
b c

) a

√
b c

Exercise: Two hallways, meeting at right angles, have widths a and b. Find the width of the
longest pole which can be slid around the corner.

Exercise: The bottom free corner of a page in a 8 inch by 6 inch book is folded back to the
binding and creased so that the crease is as short as possible. What is measure of the angle formed
by the crease at the bottom of the page?

11 Integration Techniques and Applications

11.1 Symbolic Integration Problems

You can use the student package in maple to practice your integration techniques. First load the
student package by typing

> restart;with(student):

Then read over the help screens on changevar , intparts , and value , paying particular attention
to the examples at the bottom of the screens. Here are some examples.

11.1.1 A Substitution Problem:

Integration by substitution is based on the chain rule. Thus if we have an integral which looks
like

∫

f(g(x)) ( d
dx g(x)) dx, then by make the change of variable g(x) = u, and letting du = ( d

dx g(x)) dx

we have a new, perhaps simpler integral
∫

f(u) du, to work on.
In Maple this is accomplished using the word changevar from the student package.
Find an antiderivative of
> assume(u::real,x::real);

> F := Int(1/sqrt(1+sqrt(x)),x);

F :=

∫

1
√

1 +
√

x˜
dx˜

Let’s try the change of variable sqrt(x) = u .
> G := changevar(sqrt(x)=u,F);

G :=

∫

2 u˜√
1 + u˜

du˜

This does not seem to help. Lets try 1 + sqrt(x)= u

> G := changevar(1+sqrt(x)=u,F);

G :=

∫

2
√

1 − 2 u˜ + u˜2

√
u˜

du˜

Now we can do it by inspection, so just finish it off.
> G := value(G);

G :=
4

3
signum(−1 + u˜)

√
u˜ (u˜ − 3)

Now substitute back and add in the constant.
> F := subs(u=sqrt(x),G) + C;

F :=
4

3
signum(−1 +

√
x˜) x˜(1/4) (

√
x˜ − 3) + C

Integration by substitution is the method use try after you decide you can’t find the antideriva-
tive by inspection.

11.1.2 An Integration by Parts Problem:

Integration by parts is based on the product rule for derivatives. It is usually written
∫

u dv = uv −
∫

v du.
It turns one integration problem into one which ’may’ be more doable. Once you decide to use
parts, the problem is what part of the integrand to let be u.

Integrate
> F := Int(x^2*arctan(x),x);

F :=

∫

x2 arctan(x) dx

The word is intparts. Let’s try letting u = x2 .
> G := intparts(F,x^2);

G := x2 (x arctan(x) − 1

2
ln(1 + x2)) −

∫

2x (x arctan(x) − 1

2
ln(1 + x2)) dx

That was a bad choice. Try letting u = arctan(x)
> G := intparts(F,arctan(x));

G :=
1

3
arctan(x)x3 −

∫

1

3

x3

1 + x2
dx

This is much more promising. Split off the integral on the end.

> H := op(2,G);

H := −
∫

1

3

x3

1 + x2
dx

Now do a partial fractions decomposition of the integrand of H, using parfrac.
> H:= Int(convert(integrand(H),parfrac,x),x);

H :=

∫

1

3
x − x

3 + 3x2
dx

Now we can do it by inspection.
> H1 := 1/6*x^2 - 1/3*1/2*ln(1+x^2);

H1 :=
1

6
x2 − 1

6
ln(1 + x2)

Let’s check this with the student value.
> simplify(value(H-H1));

−1

6
ln(3)

Note the difference of a constant, which is fine for antiderivatives.
ETAIL : The problem of choosing which part of the integrand to assign to u can often be solved

quickly by following the etail convention. If your integrand has an Exponential factor, choose that
for u, otherwise if it has a Trigonometric factor, let that be u, otherwise choose an Algebraic factor
for u, otherwise chose an Inverse trig function, and as a last resort choose u to be a logarithmic

factor. Let dv be what’s left over.

11.1.3 A Trig Substitution:

Find an antiderivative of
> F := Int(x^3/sqrt(x^2+1),x);

F :=

∫

x3

√
1 + x2

dx

The presence of x2 + 1 suggests letting x = tan(t).

> G := changevar(x=tan(t),F,t);

G :=

∫

tan(t)3
√

1 + tan(t)2 dt

Now use the trig identity 1 + tan(t)2 = sec(t)2 .

> G := subs(sqrt(1+tan(t)^2)=sec(t),G);

G :=

∫

tan(t)3 sec(t) dt

Another substitution into the integrand.

> G := subs(tan(t)^3 = (sec(t)^2-1)*tan(t),G);

G :=

∫

(sec(t)2 − 1) tan(t) sec(t) dt

Let’s make a change of variable,
> H := changevar(sec(t)=u,G);

H :=

∫

u2 − 1 du

From here, we can do it by inspection.
> H := value(H);

H :=
1

3
u3 − u

Now unwind the substitutions.
> G := subs(u=sec(t),H);

G :=
1

3
sec(t)3 − sec(t)

> F := subs(t = arctan(x),G);

F :=
1

3
sec(arctan(x))3 − sec(arctan(x))

> F := subs(sec(arctan(x))=sqrt(1+x^2),F) + C;

F :=
1

3
(1 + x2)(3/2) −

√
1 + x2 + C

Checking this calculation:

> F1 := int(x^3/sqrt(x^2+1),x);

F1 :=
1

3
x2

√
1 + x2 − 2

3

√
1 + x2

It looks different, but is it?

> simplify(F-F1);

C

Yes, but only by a constant.

11.1.4 A Partial Fractions Problem

Integrate the rational function
> y :=(4*x^2+x -1 )/(x^2*(x-1)*(x^2+1));

y :=
4x2 + x − 1

x2 (x − 1) (1 + x2)

First get the partial fractions decomposition of y.

> y := convert(y,parfrac,x);

y :=
1

x2
+

2

x − 1
− 3 + 2x

1 + x2

We can almost do this by inspection, except for the last term.

> F := Int(y,x);

F :=

∫

1

x2
+

2

x − 1
− 3 + 2x

1 + x2
dx

> F := expand(F);

F :=

∫

1

x2
dx + 2

∫

1

x − 1
dx − 3

∫

1

1 + x2
dx − 2

∫

x

1 + x2
dx

Now we can do each one by inspection. So we’ll just use value .
> F := value(F) + C;

F := −1

x
+ 2 ln(x − 1) − 3 arctan(x) − ln(1 + x2) + C

11.1.5 Problems:

Exercise:Use the student package to perform the following integrations.
∫

cos(x)
√

1 + sin(x)
dx

∫

3x − 7

(x − 1)2 (x − 2)3
dx

∫

x2 sin(a x) dx

∫

ln(x +
√

x) dx

∫

z5

√
z2 + 1

dz

∫

1

e(3 x) − 1
dx

∫

x arcsin(2x) dx

Exercise: Find the area of the region enclosed by the x-axis and the curve y = x sin(x) on the
interval [0, π]. Sketch the region. Then find the vertical line x = a that divides the region in half
and plot it.

Exercise: Find the length of the graph of the parabola y = x2 from O(0,0) to P(10,100). Find
the point Q(a, a2) on the graph which is 10 units from O along the graph. Make a sketch, showing
the points O, P, and Q on the graph.

Exercise: Find the volume of the solid of revolution obtained by revolving the region trapped
between the the graph of y = ex sin(x)on [0, n π]and the x-axis about the x-axis. Sketch a graph.
Does this volume approach a finite limit as n gets large?

11.2 Numerical Integration Problems

If you have a definite integral that you need to evaluate to a certain accuracy, one can always use a
numerical integration formula or ”rule”, such as the trapezoid rule or Simpson’s rule. The student
package contains words for these two rules. And you may have already discovered on your own that
when evalf(int(f(x),x=a..b); is entered, a numerical method is used to calculate an approximation.

98



11.2.1 The Trapezoid Rule

The trapezoid rule is named for the way it approximates an integral by adding the signed areas
of a number of trapezoids. The trapezpoid rule is already defined in the student package;

> student[trapezoid](f(x),x=a..b,n) ;

1

2

(b − a)

(

f(a) + 2

(

n−1
∑

i=1

f(a +
i (b − a)

n
)

)

+ f(b)

)

n
Exercise: Derive the trapezoid rule directly from the definition of the area of a trapezoid as

base (h1+h2 )
2 , where h1 and h2 are the heights of the trapezoid.

Here is a Maple procedure which draws a picture of the trapezpoids used to approximate the
integral.

> drawtrap := proc(f,a,b,n)
local dx,div, traps,i,tn,nn,ai,aim1,fai,faim1,clr;
if n > 50 then RETURN(‘Too many subdivisions..‘) fi;

> dx := evalf((b-a)/n);
div := (a,dx,i) -> evalf(a+i*dx);
traps := plot(f,a..b,thickness=2,color=black):

> for i from 1 to n do
aim1 := div(a,dx,i-1);
ai := div(a,dx,i);

> faim1 := evalf(f(aim1));
fai := evalf(f(ai)):
clr := tan;

> if faim1*fai <= 0 then clr := yellow fi;
if faim1 <= 0 and fai <= 0 then clr := aquamarine fi;
traps := traps, plots[polygonplot]([[aim1,0],

> [ai,0],[ai,fai],
[aim1,faim1]],
color=clr) od;

> tn := evalf(student[trapezoid](f(x),x=a..b,n));
### WARNING: semantics of type ‘string‘ have changed
tn := convert(tn,string);

> ### WARNING: semantics of type ‘string‘ have changed
nn := convert(n,string);
plots[display]([traps],

> title=cat(‘T‘,nn,‘ = ‘ ,tn));
end:

> drawtrap(sin,0,3*Pi,8);

Problem. Use student[trapezoid] to estimate
∫ 1
0 e(x2) dx with 10 through 50 subdivisions.

> for i from 1 to 5 do
evalf(student[trapezoid](exp(x^2),x=0..1,10*i))
od;

1.467174692

1.463783892

1.463155038

1.462934871

1.462832952

By looking at this sequence, we would feel fairly confident, (but not absolutely certain) that
the value of the integral is between 1.46 and 1.47 There is a theorem which gives a bound on the
error in the trapezoid rule .

Theorem. If there is a positive number K1 such that
∣

∣

∣

d
dx f(x) ≤ K1

∣

∣

∣ for all x in [a,b], then the

actual error
∣

∣

∣

∫ b
a f(x) dx − trapn

∣

∣

∣ is no more than K1 (b−a)2

2 n .

Now the derivative of e(x2)is 2x e(x2). and clearly the maximum of the derivative on [0,1] is
at x=1, so we can take K1 = 2 e. In order for the actual error in using the nth trapezoid rule to
estimate the integral above to be less than .01 say, the theorem guarantees that if n is chosen large
enough to satisfy the inequality
2 e 12

2 n ≤ .1e-1
. Solving this,
> solve(2*exp(1)*1^2/(2*n) <= 1/100,n);

RealRange(−∞, Open(0)), RealRange(100 e, ∞)

> evalf(100*exp(1));

271.8281828

we see that taking n = 272 will suffice, although this is much too conservative, as we can see
from our calculations with trapezoid above.

11.2.2 Problems

Exercise: Use the theorem find a value for n so that if
∫ 1
0 cos(x2 + x) dx is estimated with the

trapezoid rule with n subdivisions, the estimate is within 1/1000 of the correct value. Use trapezoid

to see if the estimate is conservative.

11.2.3 Simpson’s Rule.

Simpson’s rule is named for it’s discoverer, Thos. Simpson, in the early 1700’s. It is obtained
by interpolating a quadratic through the endpoints and midpoint of the function and using the
integral of the quadratic to estimate the integral of the function. So if the quadratic function is
called quad, then quad(a) = f(a), quad(b)=f(b) and quad((a+b)/2)=f((a+b)/2). Using this, we
can see that quad looks like

> restart;

> quad := x->f(a) + (f(b)-f(a))/(b-a)*(x-a) + d*(x-a)*(x-b);

quad := x → f(a) +
(f(b) − f(a)) (x − a)

b − a
+ d (x − a) (x − b)

and we can determine d by solving the equation
> eq := quad((a+b)/2)= f((a+b)/2);

eq := f(a) +
(f(b) − f(a)) (−1

2
a +

1

2
b)

b − a
+ d (−1

2
a +

1

2
b) (

1

2
a − 1

2
b) = f(

1

2
a +

1

2
b)

> d :=solve(eq,d);

d := −2
−f(a) − f(b) + 2 f(

1

2
a +

1

2
b)

b2 − 2 b a + a2

> quad(x);

f(a) +
(f(b) − f(a)) (x − a)

b − a
− 2

(−f(a) − f(b) + 2 f(
1

2
a +

1

2
b)) (x − a) (x − b)

b2 − 2 b a + a2

> int(quad(x),x=a..b);

−1

6

(−6 f(b) a2 − 3 f(a) b a + 3 f(b) b a + 12%1 b a − f(a) b2 − f(b) b2 − 4%1 b2) b

b2 − 2 b a + a2

− 1

6

(f(b) a2 + f(a) a2 + 4%1 a2 − 3 f(a) b a − 12%1 b a + 3 f(b) b a + 6 f(a) b2) a

b2 − 2 b a + a2

%1 := f(
1

2
a +

1

2
b)

Now simplify this and factor the result of that
> simplify(%);

−1

6
f(b) a − 1

6
f(a) a − 2

3
f(

1

2
a +

1

2
b) a +

1

6
f(b) b +

1

6
f(a) b +

2

3
f(

1

2
a +

1

2
b) b

> factor(%);

−1

6
(−b + a) (f(b) + f(a) + 4 f(

1

2
a +

1

2
b))

Voila ! Simpson’s rule. with two subintervals. Simpson’s rule with an even number n of
subintervals is defined in the student package.

> for i from 1 to 5 do evalf(student[simpson](exp(x^2),x=0..1,10*i))
od;

1.462681400

1.462653625

1.462652118

1.462651864

1.462651794

We can see that Simpson’s rule converges much more quickly on this integral than the trapezoid
rule did above. There is a theorem which govern’s the error in Simpson’s rule.

Theorem: If you can find a positive number K4 such that
∣

∣

∣

d4

dx4 f(x) ≤ K4

∣

∣

∣ for all x in [a,b], then

the actual error
∣

∣

∣

∫ b
a f(x) dx − Sn

∣

∣

∣ is no more than K4 (b−a)5

180 n4 .

We can apply this theorem to find out how large n must be taken so that the error in using
Simpson’s rule to estimate

∫ 1
0 e(x2) dx is no more than 1/100. The fourth derivative is

> diff(exp(x^2),x,x,x,x);

12 e(x2) + 48x2 e(x2) + 16x4 e(x2)

This is maximum at x=1, so we can take K4 = 76 e. Now solve the inequality for n .
> solve(76*exp(1)*1^5/(180*n^4)<= 1/100,n) ;

RealRange(−∞, −1

3
380(1/4)

√
3 (e)(1/4)), RealRange(

1

3
380(1/4)

√
3 (e)(1/4), ∞)

We can take n to be any even integer no smaller than
> evalf(1/3*380^(1/4)*3^(1/2)*exp(1)^(1/4));

3.273097124

so this is much better than the previous bound we got with the trapezoid rule.
We can check this prediction of ’4 subintervals suffice’ by using evalf(int) to compute the value

correct to 10 digits and using this as the true value.
> truevalue := evalf(int(exp(x^2),x=0..1));

truevalue := 1.462651746

> approxvalue := evalf(student[simpson](exp(x^2),x=0..1,4));

approxvalue := 1.463710760

> approxvalue - truevalue;

0.001059014

So we see that the theorem predicted correctly. Whew!
Exercise: Use the theorem find a value for n so that if

∫ 1
0 cos(x2 + x) dx is estimated with

Simpson’s rule with n subdivisions, the estimate is within 1/1000 of the correct value. Use stu-
dent[simpson] to see if the estimate is conservative. Also use evalf(int to check that the value of n
predicted by the theorem is large enough.

11.2.4 More problems with Trapezoid and Simpson

Problem: If you apply Simpson’s rule with only two subintervals to any cubic function, you will
get the exact answer every time. Verify this assertion for the cubic 5x3 − 2x2 + 5x − 3. Prove the

assertion, using the theorem giving the error bound on Simpson’s rule.
Problem: Use the error bound on Simpson’s rule above to estimate the error in using Simpson’s

rule with 6 subintervals to approximate ln(3). Use evalf(int to compute the truevalue and verify
that the estimate on the error is not smaller than the actual error.

Solution:
First, define the function and take its 4th derivative.
> f := x -> 1/x;

f := x → 1

x
> f4 := diff(f(x),x,x,x,x);

f4 :=
24

x5

The 4th derivative is clearly maximum on the left hand endpoint of the interval [1,3], so
> K[4] := 24;

K4 := 24

The bound on the error given by the theorem is
> sbound := K[4]*(3-1)^5/(180*6.^4);

sbound := 0.003292181069

So the error is less than 5 thousandths.

Exercise: The integral
∫ 1
0

√

sin(x) dx is difficult to estimate to within 1/1000 using the trapezoid
rule or Simpson’s rule. For one thing, the error bound theorems don’t apply. Why? But if we
we make the change of variable sqrt(sin(x)) = u , then we get a new integral to which the error
bounding theorems apply. Why? How many subdivisions using Simpson do you need to estimate
the new integral to within 1/1000? This exercise shows that the method of substitution is useful
in numerical integration as well as symbolic integration

> Int(sqrt(sin(x)),x=0..1) =
student[changevar](sqrt(sin(x))=u,Int(sqrt(sin(x)),x=0..1));

∫ 1

0

√

sin(x) dx =

∫

√
sin(1)

0
2

u2

√
1 − u4

du

12 Taylor’s Theorem

12.1 Taylor polynomials

Suppose the nth derivative of y = f(x) is defined at x = a . Then the nth Taylor polynomial for
f at x = a is defined as follows:

> p[n](x) = sum((D@@i)(f)(a)/i!*(x-a)^i,i=0..n);

pn(x) =
n
∑

i=0

(D(i))(f)(a) (x − a)i

i!

The Taylor remainder function is defined as Rn(x) = |f(x) − pn(x)|
There is a word, taylor, in the Maple vocabulary already which compute Taylor polynomials.

Suppose we want the 11 th Taylor polynomial of the sin function at x = 0.
> p11 := taylor(sin(x),x=0,12);

p11 := x − 1

6
x3 +

1

120
x5 − 1

5040
x7 +

1

362880
x9 − 1

39916800
x11 + O(x12)

p11 is not actually a polynomial because of the term at the end which is used to signal which
polynomial is represented (in case some of the coefficients are 0). We can convert to a polynomial.

> p11 := convert(p11,polynom);

p11 := x − 1

6
x3 +

1

120
x5 − 1

5040
x7 +

1

362880
x9 − 1

39916800
x11

The Taylor polynomials are usually good approximations to the function near a. Let’s plot the
first few polynomials for the sin function at x =0.

> sinplot := plot(sin,-Pi..2*Pi,thickness=2):
> tays:= plots[display](sinplot):

for i from 1 by 2 to 11 do
tpl := convert(taylor(sin(x), x=0,i),polynom):

> tays := tays,plots[display]([sinplot,plot(tpl,x=-Pi..2*Pi,y=-2..2,
color=black,title=convert(tpl,string))]) od:

> plots[display]([tays],view=[-Pi..2*Pi,-2..2]);

0

–2

–1

0

1

2

y

–2 2 4 6
x

Just how close the polynomials are to the function is determined using Taylor’s theorem below.

12.2 Taylor remainder theorem

Theorem: (Taylor’s remainder theorem) If the (n+1)st derivative of f is defined and bounded in

absolute value by a number M in the interval from a to x, then Rn(x) ≤ M |x−a|(n+1)

(n+1)!
This theorem is essential when you are using Taylor polynomials to approximate functions,

because it gives a way of deciding which polynomial to use. Here’s an example.
Problem Find the 2nd Taylor polynomial p[2] of f(x) = ln(x) sin(ex) + 1 at x = 1. Plot both the

polynomial and f on the interval [.5,1.5]. Determine the maximum error in using p[2] to approximate
ln(x) in this interval.

Solution:

> f := x -> ln(x)*sin(exp(x))+1;

f := x → ln(x) sin(ex) + 1

> fplot := plot(f,.5..1.5,thickness = 2):

> p[2] := x -> sum((D@@i)(f)(1.)/i!*(x-1.)^i,i=0..2);

p2 := x →
2
∑

i=0

(D(i))(f)(1.) (x − 1.)i

i!

> p[2](x);

0.5892187091 + 0.4107812909x − 2.683740378 (x − 1.)2

> t2 := unapply( convert(taylor(f(x),x=1,3),polynom),x);

t2 := x → 1 + sin(e) (x − 1) + (cos(e) e − 1

2
sin(e)) (x − 1)2

> tplot := plot(t2,1..1.5,color=black):

> plots[display]([fplot,tplot]);

0.3

0.4

0.5

0.6

0.7

0.8

0.9

1

0.6 0.8 1 1.2 1.4

In order to use Taylor’s remainder theorem, we need to find a bound M on the 3rd derivative
of the function f. In this case, we could just plot the third derivative and eyeball an appropriate
value for M.

> plot((D@@3)(f),.5..1.5) ;

We could use M = 75.
> M := 75;

M := 75

So the remainder R2 = |f(x) − p2(x)| is bounded by

> M/3!*(1.5-1)^3;

1.562500001

We can see from the plot of f and the polynomial that the actual error is never more than about
.1 on the interval [.5,1.5].

Another example:

Which Taylor polynomial would you use to approximate the sin function on the interval from
-Pi to Pi to within 1/10ˆ6?

Solution:
Well, 1 is a bound on any derivative of the sin on any interval. So we need to solve the inequality
> ineq := 1/n!*Pi^n <= 1/10^6;

ineq :=
πn

n!
≤ 1

1000000
for n. Solve will not be much help here because of the factorial, but we can find the smallest n

by running through a loop.
> n := 1: while evalf(1/n!*Pi^n) > 1/10^6 do n := n+1 od: print

(‘take n to be ‘,n);

take n to be , 17

> (seq(evalf( 1/n!*Pi^n) ,n=15..20));

0.00002191535349, 0.4303069596 10−5 , 0.7952054018 10−6 , 0.1387895250 10−6 ,

0.2294842906 10−7 , 0.3604730807 10−8

> restart;

> t17 := convert(taylor(sin(x),x=0,18),polynom);

t17 := x − 1

6
x3 +

1

120
x5 − 1

5040
x7 +

1

362880
x9 − 1

39916800
x11 +

1

6227020800
x13

− 1

1307674368000
x15 +

1

355687428096000
x17

> plot(t17,x=-Pi..Pi);

Looks pretty much like the sin function.

12.3 Problems

Exercise: Show that cos(x) is approximated to within 7 decimals by 1 − x2

2! + x4

4! − x6

6! + x8

8! for all
x in [−π

4 , π
4 ] .

Exercise: Use taylor and convert(..,polynom) to compute and plot, on the interval specified, the
first few taylor polynomials of the following functions. Observe the convergence of the polynomials
to the function and make comments.
f(x) = ln(x)

at x=1, on the interval [-1,3].
f(x) = 1

1−x
at x =0 on the interval [-2,2]

f(x) = arctan(x)
at x = 0 on the interval [-2..2]

Exercise: Write a procedure to compute sin(x) for any x by using p[5]. restricted to the interval
[0,Pi/4].

Outline of solution: If x is negative, replace x with −x and use the oddness sin(x) = −sin(−x)
property. If x is greater than or equal to 2*Pi, then replace x with x-2*Pi and use the periodicity
sin(x) = sin(x − 2π) . Repeat this step until [0, 2*Pi ). If Pi/4 < x < Pi/2 , then use the trig
indentity sin(x) = 1 − sin(π

2 − x)
and approximate sin(π

2 − x) by p5(
π
2 − x). If π

2 < x and x < π, then sin(x) = sin(π − x). If
π < x and x < 2π, then sin(x) = −sin(x − π).

Exercise: Find the smallest n such that the nth Taylor polynomial p[n](x) for y = exat x = 0
approximates exp(x) to within 10(−12)for x in [0,1]. (You will want to set Digits equal to 15 in
order to do this one.

or so to do this problem.)

13 Sequences and Series

13.1 Sequences

We have used sequences lots of times before. The sequence of estimates to the solution of an
equation generated by Newton’s Method is one. The sequence of estimates to the integral of a
function over an interval obtained by subdividing the interval into more and more subintervals
is another. These are examples of potentially infinite sequences. Theses are sequences we hope
converge to the answer we seek, whether it be the solution of an equation or the value of an
integral.

Formally, a sequence of numbers is defined as a function f whose domain is the positive
integers. The terms of the sequence are the values of the function. So for example the 10th term
of the sequence f is f(10).

A sequence f converges to a limit L if each interval containing L contains all but finitely
many terms of the sequence. In this case, we would write limn→∞ f(n) = L.

The Maple word limit can be used to calculate many limits of sequences in a painfree manner.
For example,

> limit((1+1/n)^n,n=infinity);

e

The next theorems summarize many of the properties of convergent sequences.
Theorem: If an is an increasing sequence (ie, ai ≤ ai+1 for all i), then an converges if there is

an upper bound on the terms of an.

Theorem: If an converges to L and bn converges to M , then an bn converges to LM , an + bn

converges to L + M , and an − bn converges to L − M . Also, if M 6= 0, then an
bn

converges to L
M .

Theorem: If an is a sequence of positive numbers and bn is a sequence which converges to 0,
then if an ≤ bn for all n , then an converges to 0.

Theorem: If f is continuous at x = L, and an is a sequence converging to L, then the sequence
f(an) converges to f(L).

13.1.1 Periodic Points of functions.

Let f be a function, and let a be a point in the domain of f. If each value of f is in the domain of
f, we can generate the sequence of iterates of a under f as follows: a1 = f(a), a2 = f(a1), and
in general an+1 = f(an) for each positive integer n. If n is a positive integer such that an = a, but
ai 6= a for for all positive integers i < n, then a is called a periodic point of period n for f.

Period one points are called fixed points . You can locate the fixed points of a function by
looking to see where the graphs of y = f(x)and y = x cross. For example, the cosine function has
one fixed point, as we can see by plotting.

> plot({cos(x),x},x=-Pi..Pi);

To find the fixed point more precisely, use fsolve .

> fix := fsolve(cos(x)=x,x,0..Pi);

fix := 0.7390851332

An attracting fixed point is a fixed point a with the property that for points b close to a,
> limit(b[n],n=infinity) = a;

lim
n→∞

bn = a

where b1 = b , and bn = f(bn−1) for n = 2, 3 ...
A repelling fixed point is a fixed point a with the property that for points b close (but not

equal) to a,

> limit(b[n],n=infinity) <> a;

lim
n→∞

bn 6= a

where b1 = b , and bn = f(bn−1) for n = 2, 3, ... .
˜
Here is a simple procedure to investigate periodic points and fixed points of a function.

> iterate := proc(f,n,x)
local a,i,s;
a := evalf(x);

> s := a;
for i to n do a := f(a);
s := s,a od
end:

For example, to investigate whether the fixed point of the cosine function is attracting or not,
we can iterate the function at a point near the fixed point. Using the fixed point of the cos function,

> fix := fsolve(cos(x)=x,x);

fix := 0.7390851332

> iterate(cos,10,fix-1),fix;

−0.2609148668, 0.9661543793, 0.5684675409, 0.8427269503, 0.6654297419,

0.7866515363, 0.7062199575, 0.7608204115, 0.7242705678, 0.7489829351,

0.7323817612, 0.7390851332

The fixed point seems to be an attracting one. On the other hand if we look at the fixed points
of 2x(1-x),

> f := x-> 2*x*(1-x);

f := x → 2x (1 − x)

> fix := fsolve(f(x)=x,x);

fix := 0, 0.5000000000

> iterate(f,10,fix[1]-.2);

−0.2, −0.48, −1.4208, −6.87894528, −108.3976669, −23716.90372, −0.1125030478 1010 ,

−0.2531387156 1019 , −0.1281584187 1038 , −0.3284916056 1075 ,

−0.2158134698 10150

> iterate(f,10,fix[1]+.1);

0.1, 0.18, 0.2952, 0.41611392, 0.4859262512, 0.4996038592, 0.4999996862,

0.5000000000, 0.5000000000, 0.5000000000, 0.5000000000
> iterate(f,10,fix[2]+.4);

0.9000000000, 0.1800000000, 0.2952000000, 0.4161139200, 0.4859262512,

0.4996038592, 0.4999996862, 0.5000000000, 0.5000000000, 0.5000000000,

0.5000000000
It seems that 0 is an repelling fixed point and that .5 is an attracting fixed point. Let’s define a

visual word to go with iterate. We have added a domain and range to allow you to determine the
viewing window.

> viterate := proc(f,n,start,domain,range)
local a, i, s, gra, gpl, fpl, ipl;
a := evalf(start);

> gra := [a,f(a)];
for i to n do a := f(a);
gra := gra,[a,a],[a,f(a)];

> od:
gpl := plot([gra],color=red);
fpl := plot(f,domain,color=black);

> ipl := plot(x->x,domain,color=blue);
print(plots[display]([gpl,fpl,ipl],view=[domain,range]));
end:

> viterate(x->2*x*(1-x),10,.8,-1..1 ,-1..1);

This gives a nice visual tool to investigate fixed points and periodic points of functions.

13.1.2 Problems

Exercise: Use iterate or viterate to check more starting points close the the fixed point of cos. Do
you remain convinced that it is a repelling fixed point?

> fx := fsolve(cos(x)=x,x);

fx := 0.7390851332

> viterate(cos,10,.1,-1..2,-1..1);

2. Find the periodic points of period 2 of y = 2x (1 − x). . (Hint: the period points of order
two of f would be the fixed points of f (2) which are not fixed points of f. Classify them as repelling,
attracting, or neither.

Exercise: Let an be a sequence of positive numbers converging to 0. Imagine yourself starting
at the origin and travelling east a1 miles, then turning north and going a2 miles, then west a3 miles,
and so forth, cycling through the directions as you go through the sequence an. (1) Where do you
end up? (2) How far do you travel along your path. Work the answers out for the sequences 1

n and
1
2n .

112



Solution:
Call the point where we end up [x,y]. Then x = a1 − a3 + a5 − a7 ..., the sum of the alternating

series of odd terms of the sequence
an

and y is the sum of the alternating series of even terms of the sequene. So for the sequence
> a := n-> 1/n;

a := n → 1

n
> x = sum((-1)^n*a(2*n+1),n=0..infinity);

x =
1

4
π

> y = sum((-1)^n*a(2*(n+1)),n=0..infinity);

y =
1

2
ln(2)

and for the sequence
> a := n-> 1/2^n;

a := n → 1

2n

> x = sum((-1)^n*a(2*n+1),n=0..infinity);

x =
2

5
> y = sum((-1)^n*a(2*(n+1)),n=0..infinity);

y =
1

5
(2) The total distance we travel along the path is the sum of all the distances travelled. So for

the sequence
> a := n->1/n;

a := n → 1

n
> ’distance’ = sum(a(n),n=1..infinity);

distance = ∞
the distance is infinity! This is perhaps surprizing at first, since each time we turn we go a

smaller distance than the last time. On the other hand, for the sequence
> a := n->1/2^n;

a := n → 1

2n

> ’distance’ = sum(a(n),n=1..infinity);

distance = 1

The distance travelled is only 1 unit.

Exercise: Suppose we wanted to draw the paths described in the above problem. Here is a
procedure which will do that. Use it to draw the paths for the sequences an = 1

n , and an = 1
2n .

> cycle := proc(a,m)
local path,i, dir,x,y,pt,ed;
x := evalf(sum((-1)^n*a(2*n+1),n=0..infinity));

> y := evalf(sum((-1)^n*a(2*(n+1)),n=0..infinity));
path := [0,0];
dir := 1,0;

> for i from 1 to m do
path := path,[path[i][1]+dir[1]*a(i),
path[i][2]+dir[2]*a(i)];

> dir := -dir[2],dir[1];
od;
pt := plot([path],scaling=constrained,color=red,thickness=2);

> ed := plot({[x,y]},style=point,symbol=box):
plots[display]([pt,ed],title=cat("end at ",convert([x,y],string)));
end:

Solution:
For the sequenc 1/n
> cycle(n->1/n,15);

end at [.7853981635, .3465735903]

> cycle(n->1/2^n,15);

end at [.4000000000, .2000000000]

13.2 Series

Definition: A series
∑∞

n=1 an consists of two sequences: The sequence an of terms of the series and
the sequence Sn =

∑n
i=1 ai of partial sums of the series. If the sequence of partial sums converges

to a limit L, then the series is said to converge to L and we write
∑∞

n=1 an = L.
Suppose you have established somehow, either directly or by some test, that a series converges

to a number L. How do we calculate this number to any specified accuracy?

Not surprizingly, Maple can sum a lot of series already. For example, the sum of a geometric
series is easy for Maple to compute.

> restart;

> Sum(a*r^n,n=0..infinity)=sum(a*r^n,n=0..infinity);

∞
∑

n=0

a rn = − a

r − 1

Maple also knows that the harmonic series diverges to infinity.
> sum(1/n,n=1..infinity);

∞
It also knows how to compute the sums of the various convergent p-series. For example, the

3-series sums to
> Sum(1/n^3,n=1..infinity)=sum(1/n^3,n=1..infinity);

∞
∑

n=1

1

n3
= ζ(3)

The Riemann Zeta function is defined for all p > 1 to give the sum of the p-series.
So, for example, the sum of the 3 series is
> Zeta(3)=Zeta(3.);

ζ(3) = 1.202056903

If the series converges fast enough you can look at the sequence of partial sums and get the
desired accuracy.

Lets see how fast the 3-series converges to Zeta(3).
> for n from 10 by 100 to 300 do

Sum(1/i^3,i=1..n)=evalf(sum(1/i^3,i=1..n)) od;

10
∑

i=1

1

i3
= 1.197531986

110
∑

i=1

1

i3
= 1.202015955

210
∑

i=1

1

i3
= 1.202045619

Well, the sum of the first 100 terms is accurate to 4 significant figures.

You can’t decide for sure by looking at first few partial sums of a series that the series converges.
For example, look at a few partial sums of the harmonic series.

> seq(evalf(sum(1/i,i=1..100*n)),n=1..5); n:=’n’:

5.187377518, 5.878030948, 6.282663880, 6.569929691, 6.792823430

Hmmm. You can’t really tell by looking at these that the harmonic series doesn’t converge.

13.2.1 Problems

Exercises: In each of the problems below, determine whether the series converges or diverges. Give
a reason in each case. For the convergent series, get an estimate correct to 2 decimal places of the
sum of the series using psums or some other word of your own devising. You can check with sum
to see if Maple can sum it.
∑∞

n=1
1

(3+2 n)2

This series converges by comparison with the p-series
∑∞

n=1
1
n2 . Each partial sum is bounded

above by ζ(2) and the partial sums form an increasing sequence, so we know the sequence of partial
sums converge. Checking to see what is programmed into Maple,

> sum(1/(3+2*n)^2,n=1..infinity)=
evalf(sum(1/(3+2*n)^2,n=1..infinity));

−10

9
+

1

8
π2 = 0.122589440

we get an exact sum. Check a few partial sums .
> seq(evalf(sum(1/(3+2*i)^2,i=1..100*n)),

n=1..5); n:=’n’:

0.1201384783, 0.1213518178, 0.1217616252, 0.1219675488, 0.1220914312

∑∞
n=2

1
n ln(n)2

The function 1
n ln(n)2 is a decreasing for n > 1 (Take the derivative), so we can use the integral

test on this one.
> int(1/(n*(ln (n))^2),n=1..infinity);

∞
Since the integral diverges, the series diverges.

∑∞
n=0

1

(2 n+1)(
1
3 )

This series diverges, since it is a p-series with p < 1

∑∞
n=0

1+2n

1+3n

∑∞
n=0

n5+4 n3+1
2 n9+n4+2

This series converges by comparison with
∑ n6

n9 .
> evalf(int(( n^5+4*n^3+1)/

(2*n^9+n^4+2 ),n=1..infinity));

0.4309605111

∑∞
n=1 sin( 1

n4 )
This series converges by comparison with the 4-series.
> evalf(sum(sin(1/(n^4)),n=1..infinity));

0.9237532120

∑∞
n=1 n e(−n2)

13.3 Two interesting curves

13.3.1 The Snowflake Curve

The Snowflake Curve was initially described by Koch as an affirmative solution to the problem
of whether there is a continuous curve that has no tangent line at any point on the curve.

It is defined as the limit of the sequence of curves generated by the procedure snowflake given
below. Type in the words defined below and then enter snow(4) to get a feel for what the snowflake
curve looks like. The first word performs a basic operation on any segment, which we think of as a
list of its endpoints. It replaces the segment of length d with a sequence of four segments of length
d/3 obtained in the following way: Start at the left endpoint of the segment and go d/3 of the way
along the segment, turn left 60 degrees and go the same distance, turn right 60 degrees and go the
same distance, then turn left 60 degrees and proceed d/3 units along the original segment to the
right endpoint. Note that the resulting path has three points where there is no tangent.

> basic := proc(p1,p2)
local dx,dy, p3,p4,p5;
dx := (p2[1]-p1[1])/3.;

> dy := (p2[2]-p1[2])/3.;
p3 := [p1[1]+dx,p1[2]+dy];
p4 := [p1[1]+2*dx,p1[2]+2*dy];

> p5 := [p1[1]+1.5*dx-sqrt(3.)/2*dy,
p1[2]+1.5*dy+sqrt(3.)/2.*dx];
p3,p5,p4,p2;
end:

So for example, if we feed the points [0,0], [1,0] into basic, we get out the following sequence of
numbers:

> basic([0,0],[1,0]);

[0.3333333333, 0], [0.5000000000, 0.2886751347], [0.6666666666, 0], [1, 0]

Notice the left endpoint of the original segment is missing from this list. That is for programming
reasons which become clear in the definition of the word flake below. The word flake takes a list of
points, which represents a sequence of line segments laid end to end, and returns a list representing
4 times as many line segments, where each segment in the original list has been replaced by the 4
segments returned by basic.

> flake := proc(fl)
local i,curve;
curve := fl[1];

> for i from 1 to nops(fl)-1 do
curve := curve ,basic(fl[i],fl[i+1] ) ;
od
end:

Now the starting point for the snowflake curve is the equilateral triangle.

> curve := [[0,0],[1/2,1/2*sqrt(3)],[1,0],[0,0]];

curve := [[0, 0], [
1

2
,

1

2

√
3], [1, 0], [0, 0]]

To draw the second stage of the snowflake curve,
> plot([flake(curve)],scaling=constrained);

Now we can define the nth stage of the snowflake curve.
> snowflake := proc(n)

local i,curve,ti;
curve := [[0,0],[1/2,1/2*sqrt(3)],[1,0],[0,0]];

> for i from 2 to n do curve := [flake(curve)] od;
ti := cat(n,‘th stage of the snowflake curve‘);
plot(curve,color=black,style = LINE,

> axes = NONE,scaling = CONSTRAINED,title = ti)
end:

> snowflake(4);

4th stage of the snowflake curve

Here’s an animation of the snowflake curve ’growing in place’. About 5 frames is all you can
usefully see.

> plots[display]([seq(snowflake(i),i=1..5)],insequence=true,axes=none,s
caling=constrained);

1th stage of the snowflake curve

Problem: What is the length of the snowflake curve?
Solution: To calculate the length of the snowflake curve we see that the first stage has length

s1 = 3 units; the 2nd stage has length s2 = s1 + s1
3 , or 4 s1

3 ; in general the nth stage has length

sn = s(n − 1) + 1 s(n−1)
3 , or (4

3)(n−1) s1 . So the length of the nth stage goes to infinity as n gets
large. This says the snowflake curve is infinitely long.

˜

13.3.2 A Spacefiller

David Hilbert in 1891 described a continuous curve whose range is the unit square! This was
contrary to the intuition of the time, which was that the range of a path had to be 1-dimensional.
Here is a Maple word that draws an approximation to Hilbert’s space filling curve . The approach
to defining the drawing procedure peano (below) is similar to that used to define snowflake (above).

> basic := proc(p1,p2,p3)
local dx,dy,p4,p5,p6,p7,p8,p9;
p4 := .5*(p1+p2);

> p9 := .5*(p2+p3);
p5 := p4+(p2-p9);
p6 := p2+(p2-p9);

> p7 := p2+(p4-p1);
p8 := p9+(p4-p1);
p4,p5,p6, p2 ,p7,p8,p9, p3 ;
end:

> peano := proc(fl)
local i,cur;
cur := [fl[1] ] ;

> for i from 1 by 2 to nops(fl)-2 do
cur := [op(cur),basic( fl[i],fl[i+1],fl[i+2] )]
od;

> end:

> fl := [[0,0],[1,1],[2,0]];

119



fl := [[0, 0], [1, 1], [2, 0]]
> for i from 1 to 4 do fl := peano(fl) od:

ti := cat(4,‘th stage of Peano’s curve‘);
plot(fl,style=LINE,axes=NONE,title=ti,
scaling=CONSTRAINED);

ti := 4th stage of Peano ′s curve

4th stage of Peano’s curve

14 Differential equations

14.1 Terminology

The simplest differential equation is of the form d
dx y(x) = f(x), where f is some known function

of x and y is an unknown function of x. These are solved by antidifferentiation. We use int. For
example, to solve the differential equation y’(x)= xˆ2*cos(x), we would type

> y := int(x^2*cos(x),x) + C;

y := x2 sin(x) − 2 sin(x) + 2x cos(x) + C

The constant of integration C can be determined once you know a single value of the function.
So if y(3) = 10, then we can get C by solving an equation.

> Csol := solve(subs(x=3,y=10),{C});
Csol := {C = −7 sin(3) − 6 cos(3) + 10}

> y := subs(Csol,y);

y := x2 sin(x) − 2 sin(x) + 2x cos(x) − 7 sin(3) − 6 cos(3) + 10

> plot(y,x=-6..6);

The theorem which tells us about uniqueness of solutions is the following.
Big theorem: If two functions have the same derivative on an interval, they differ by a constant.
A differential equation is an equation which expresses a relation between an unknown

function y and one or more of its derivatives. A solution to the differential equation is a function
which satisfies the equation.

An Initial Value Problem or IVP is a differential equation together with some initial conditions.
A solution to the IVP is a function which satisfies the equation and also the intial conditions.

The order of a differential equation is n where n is the highest order derivative appearing
in the equation.

We will concentrateon first order equations of the form y’ = f(x,y). Many interesting problems
lead to a first order equation of some sort.

The Maple word dsolve takes as input a differential equation and possibly some initial values,
and attempts to return a solution to the equation.

To illustrate how dsolve works, use it to solve the above differential equation.
> restart;

Set up the differential equation.
> diffeq := diff(y(x),x)=x^2*cos(x);

diffeq := d
dx y(x) = x2 cos(x)

use dsolve on it.
> dsolve(diffeq,y(x));

y(x) = x2 sin(x) − 2 sin(x) + 2x cos(x) + C1

Note the constant of integration. If we include the initial conditions, dsolve will determine this
constant.

> inits := y(3) = 10;

inits := y(3) = 10

> sol := dsolve({diffeq,inits},y(x));
sol := y(x) = x2 sin(x) − 2 sin(x) + 2x cos(x) − 7 sin(3) − 6 cos(3) + 10

Now to plot sol we can just plot the right hand side (rhs) of the solution:
> plot(rhs(sol),x=-6..6);

or we could turn sol into a function with unapply.

> f := unapply(rhs(sol),x);

f := x → x2 sin(x) − 2 sin(x) + 2x cos(x) − 7 sin(3) − 6 cos(3) + 10

There is a useful package of words to use called DEtools.
> with(DEtools);

[DEnormal , DEplot , DEplot3d , Dchangevar , PDEchangecoords , PDEplot , autonomous ,

convertAlg , convertsys , dfieldplot , indicialeq , phaseportrait , reduceOrder ,

regularsp , translate , untranslate , varparam ]
The word dfieldplot is one which you can use to examine qualitatively the solution curves to a

differential equation y’ = f(x,y). For example for the diffeq above,
> dfieldplot(diffeq,[y(x)],x=-6..6,y=-10..40,title=‘The direction field

of y’ = x^2*cos(x).‘);

The direction field of y’ = x^2*cos(x).

By looking at the direction field plot, you can visually sketch in the solution curves to the
differential equation. Another term for solution curve is integral curve. An isocline is a curve
that connects points where the slope is constant. For example, the isoclines of the differential
equation above are the vertical lines. When you are forced to construct a direction field plot by
hand, it is often easier to first sketch in a few isoclines before drawing in the slope vectors.

14.2 Problems leading to first order equations

A Salt tank problem
Setting: A tank initially contains 50 gallons of fresh water. Brine containing 1/4 lb salt/gal

comes into the tank at 3 gals /minute and the solution is kept homogeneous by vigorous stirring.
The mixture drains out of the tank at 3 gals /minute, so that there is always 50 gallons of solution
in the tank.

Even before we think of a differential equation, we can make a rough sketch of the amount of
salt in the tank over time. Let A(t) be the amount of salt in the tank at time t. Then A(0) = 0,
and as t increases towards infinity A(t) will increase towards 50*1/4 lbs.

Now we don’t know A(t) explicitly, but we can say what the rate of change of A(t) with respect
to t is, and this will be the differential equation we need to solve. In words, the rate of change
of A(t) is the rate at which the salt is coming in minus the rate at which it is going out. The in
rate is constant, at 3*1/4 lbs per minute. The out rate is increasing: At any particular time t, the
concentration of the salt in the tank is A(t)/50 lbs per gallon, so the salt is going out at 3* A(t)/50
lbs per minute. This give the differential equation which governs the amount of salt in the tank.

> diffeq := diff(A(t),t) = 3/4 - 3*A(t)/50;

diffeq := d
dt A(t) =

3

4
− 3

50
A(t)

The initial value of A(t) is A(0)=0, so we should get a unique solution from dsolve.
> ### WARNING: ‘dsolve‘ has been extensively rewritten, many new result

forms can occur and options are slightly different, see help page for
details
sol := dsolve({diffeq,A(0)=0},A(t));

sol := A(t) =
25

2
− 25

2
e(−3/50 t)

> plot(rhs(sol),t=0..100);

We can make a function from sol and use it to answer any question about the setting.

> A := unapply(rhs(sol),t);

A := t → 25

2
− 25

2
e(−3/50 t)

For example, what is the amount of salt in the tank after 1 minute? Answer:
> A(1.),‘lbs‘;

0.72794333, lbs

How long does it take the amount of salt in the tank to get to 10 lbs? Answer:

> solve(A(t)=10,t);

50

3
ln(5)

> evalf(%),‘minutes‘;

26.82396521, minutes

We can solve the differential equation coming out of the salt tank problem ’by hand’, because
it is separable equation . That is, it is of the form y’ = f(x)*g(y). Separable equations always
reduce to finding two antiderivatives. Just rewrite the equation in the form y’/g(y) = f(x) and
antidifferentiate both sides. If you are successful, you will have an equation in x and y(x) only.
This equation implicitly defines the solutions to your separable equation. Sometimes you can solve
the equation for y(x). Then you have found an explicit solution to the separable equation. In our
case,

> restart;

> diffeq := diff(A(t),t) = 3/4 - 3*A(t)/50;

diffeq := d
dt A(t) =

3

4
− 3

50
A(t)

We rewrite this in the A’ = 3/4-3/50*A=f(t)g(A), where f(t) = 1 and g(A) = 3
4 − 3 A

50 . Now an
antiderivative of f(t) = 1 is t + C, and an antiderivative of A’/g(A), we can get by substitution.
The equation we get after integrating both sides is

> eq := int(1/(3/4-3/50*A),A)= int(1,t)+C;

eq := −50

3
ln(

3

4
− 3

50
A) = t + C

To determine the constant of integration here, use the given initial value of A=0 when t=0.
> Csol := solve(subs({A=0,t=0},eq),{C});

Csol := {C = −50

3
ln(

3

4
)}

> assign(Csol);

Now solve for A and make it into a function.
> A := unapply(solve(eq,A),t);

A := t → 25

2

e(3/50 t) − 1

e(3/50 t)

Look ma, no dsolve!
Exercise: Suppose the salt tank develops a leak at time t=0, from which an additional 1/2 gallon

of solution per minute leaves the tank. How do you think this will affect the function A(t)? Make
a sketch to accompany your discussion. Write down the new differential equation which governs
the setting. Is it separable? Solve it with dsolve and plot over the appropriate time interval. What
is the maximum amount of salt in the tank and when is it there?

Radioactive decay: Carbon 14 dating
Problem: A piece of charcoal found at Stone Henge was determined to contain Carbon 14 in

a concentration which produced 8.2 disintegrations per minute per gram. Charcoal from a living
tree is known to produce 13.5 disintegrations per minute per gram. The half life of Carbon 14
is known to be 5568 years. Assuming that the tree was burned during the construction of Stone
Henge, estimate the age of Stone Henge.

A Solution:
The law of radioactive decay was discovered by experiment. It is simply that radioactive

materials decay (or turn into non-radioactive materials) at a rate which is proportional to the
amount of radioactive material present. We set up the differential equation so that the decay
constant, k, is positive.

> diffeq := diff(decay(t),t) = -k*decay(t);

diffeq := d
dt decay(t) = −k decay(t)

A gram of the charcoal back when Stone was being built (t=0) had enough carbon 14 to provide
13.5 disintegrations per minute, and now at t=T has enough to provide only 8.2 disintegrations
per minute. We will use the number of disintegrations per minute as our measure of the amount
of carbon 14 in the gram of charcoal. With that understood, the initial condition then is

> conds := decay(0)=13.5;

conds := decay(0) = 13.5

We could solve this by hand, but for the moment use dsolve.
> ### WARNING: ‘dsolve‘ has been extensively rewritten, many new result

forms can occur and options are slightly different, see help page for
details
soln := dsolve({diffeq,conds},decay(t));

soln := decay(t) = 13.50000000 e(−k t)

Make a function of the right hand side.
> decay := unapply(rhs(soln),t);

decay := t → 13.50000000 e(−k t)

Now determine the decay constant k. This we can do by using the half life information. Every
5568 years, the amount of carbon 14 in a gram of the charcoal is cut in half (and so the number of

disintegrations per minute is cut in half).
> k := solve(decay(5568)=1/2*decay(0),k);

k := 0.0001244876402

Now we can use the measurement of 8.2 disintegrations per minute per gram to estimate the
age of Stone Henge.

> age_of_Stone_Henge := solve(decay(t)=8.2,t);

age of Stone Henge := 4004.859682

So, Stone Henge was built right about the time the world was made (according to Bishop Usher).
Exercise: Solve the differential equation in the problem above by hand, using the fact that it is

separable.

14.3 Logistic Growth

Logistic growth occurs when the rate of change of a population is proportional to the product
of the population present and the amount of room left. This is a differential equation. k is the
constant of proportionality and C is the least upper bound on the population.

> deq := diff(p(t),t)=k*p(t)*(C-p(t));

deq := d
dt p(t) = k p(t) (C − p(t))

This equation is separable and can be solved symbolically.
> ### WARNING: ‘dsolve‘ has been extensively rewritten, many new result

forms can occur and options are slightly different, see help page for
details
sol := dsolve(deq,p(t));

sol :=
1

p(t)
=

1 + e(−k C t) C1 C

C
Seems like a strange way to write the solution. Lets turn it right side up.
Also, we can replace the constant C1*C with C1 no problem.

> p := t-> C/(1+C1*exp(-k*C*t));

p := t → C

1 + C1 e(−k C t)

By looking at the population function, we can see that as t goes to infinity p(t) goes to C. If
we are going to model a population with the logistic equation, we need to know the population at
three times, in order to get three equations to determine the three parameters k, C, and C1 in the
population function.

Suppose you measured the population at three times and got p(0) = 10000, p(10) = 20000, and
p(20) = 30000. Looks linear doesn’t it? What logistic curve fits this data?

Set up the equations.
> eq1 := p(0) = 10000;

eq1 :=
C

1 + C1
= 10000

> eq2 := p(10) = 20000;

eq2 :=
C

1 + C1 e(−10 k C)
= 20000

> eq3 := p(20) = 30000;

eq3 :=
C

1 + C1 e(−20 k C)
= 30000

Then solve for the parameters.
> sln :=solve({eq1,eq2,eq3},{C,C1,k});

sln := {k =
1

400000
ln(3), C = 40000, C1 = 3}

Make a function out of the solution.

> f := unapply(subs(sln,p(t)),t);

f := t → 40000

1 + 3 e(−1/10 ln(3) t)

Draw a picture of the population curve.
> plot([40000,f],0..70,color=black);

Where is the inflection point? In general, it will be at
> solve(diff(p(t),t,t),t);

−
ln(

1

C1
)

k C
For our particular data, it will be at
> solve(diff(f(t),t,t),t);

10

Question: What happens to the population curve as the population at t = 20 changes?
Introduce a parameter p20 to stand for the population at t=20 and resolve.

> eq3 := p(20) = p20;

eq3 :=
C

1 + C1 e(−20 k C)
= p20

> sln := solve({eq1,eq2,eq3},{C,C1,k});

sln := {C = − 400000000

p20 − 40000
,

k =
1

4000000000
ln(

−20000 + p20

p20
) p20 − 1

100000
ln(

−20000 + p20

p20
),

C1 = − p20

p20 − 40000
}

By inspection, we can see that p20 must be greater than 20000 and less than 40000.
Make a function again, this time with two variables, t and p20
> f := unapply(subs(sln,p(t)),t,p20);

f := (t, p20 ) → − 400000000

(p20 − 40000)













1 − p20 e

(

400000000
(1/4000000000 ln(

−20000+p20
p20

) p20−1/100000 ln(

p20
)) t

p20−40000

)

Now we can animate the change in the population curve as the population at t=20 runs through
its possible values. Unfortunately, you can’t see the animation if you are looking at a piece of paper.

> plots[animate](f(t,p20),t=0..150,p20=21000..39000);

